
ФЕДЕРАЛЬНОЕ АГЕНТСТВО ПО ОБРАЗОВАНИЮ 
Государственное образовательное учреждение высшего профессионального образования 

"ТОМСКИЙ ПОЛИТЕХНИЧЕСКИЙ УНИВЕРСИТЕТ" 
___________________________________________________________________________________________ 

 
 
 
 
 
 
 
 
 
 
 
 

Г.В. Кузнецов, М.А. Шеремет 
 
 

РАЗНОСТНЫЕ МЕТОДЫ РЕШЕНИЯ ЗАДАЧ 
ТЕПЛОПРОВОДНОСТИ 

 
Учебное пособие 

 
 
 
 
 
 
 
 
 
 
 
 

 
 

Издательство  
Томского политехнического университета 

Томск 2007 

DiAmon
New Stamp

http://www.c-o-k.ru/biblio


ББК 22.36:22.19я73 
УДК 536.2:519.876(075.8) 
         К 891 
 
Кузнецов Г.В., Шеремет М.А. 

К 891 Разностные методы решения задач теплопроводности: учебное 
пособие. / Г.В. Кузнецов, М.А. Шеремет. – Томск: Изд-во ТПУ, 
2007. – 172 с. 
 
 
 

В пособии приведены основные сведения по методам, алгоритмам и 
программам решения на ЭВМ основных задач теплопроводности. 
Приведены примеры решения линейных, нелинейных, одномерных, 
двумерных, нестационарных задач. Пособие предназначено для студентов 
всех специальностей и аспирантов теплоэнергетического факультета 
Томского политехнического университета. 

 
УДК 536.2:519.876(075.8) 

 
 
 

Рекомендовано к печати Редакционно-издательским советом  
Томского политехнического университета 

 
Рецензенты 

 
Доктор физико-математических наук, профессор ТГУ 

А.В. Герасимов 
 

Доктор физико-математических наук, профессор ТГАСУ 
В.Н. Барашков 

 
 
 
 
 
 

 
 

© Томский политехнический университет, 2007 
© Оформление. Издательство ТПУ, 2007 



 3

О Г Л А В Л Е Н И Е 

Введение 4 
1. Метод конечных разностей (МКР) 8 
2. Линейные задачи теплопроводности 8 
    2.1. Одномерное уравнение теплопроводности 8 
    2.2. Конечно-разностная аппроксимация граничных условий 
           второго и третьего рода 

 
30 

    2.3. Примеры краевых задач с различными граничными 
           условиями 

 
35 

    2.4. Двухслойная пластина 47 
    2.5. Задача теплопроводности с внутренними источниками 57 
    2.6. Двумерная задача теплопроводности для однородного 
           тела 

 
64 

    2.7. Двумерная задача теплопроводности для неоднородного 
           тела 

 
73 

3. Нелинейные задачи теплопроводности 89 
    3.1. Одномерное уравнение теплопроводности с зависящим 
           от температуры коэффициентом теплопроводности 

 
89 

    3.2. Одномерное уравнение теплопроводности с 
           нелинейными граничными условиями (излучение на 
           границе) 

 
 

101 
    3.3. Одномерное уравнение теплопроводности с фазовым 
           переходом на границе (испарение материала) 

 
111 

    3.4. Одномерное уравнение теплопроводности с химической 
           реакцией в материале (термическое разложение) 

 
122 

    3.5. Одномерное уравнение теплопроводности с подвижной 
           границей (промерзание влажного грунта) 

 
128 

    3.6. Двумерное уравнение теплопроводности с излучением на 
           границах 

 
140 

    3.7. Двумерное уравнение теплопроводности с фазовым 
           переходом на одной из границ 

 
150 

    3.8. Квазитрехмерная модель (излучение и конвекция по 
           третьей координате) 

 
159 

Список литературы 171 
 

 

 

 



 4

ВВЕДЕНИЕ 
 

 Изучение процессов теплообмена всегда играло видную роль в 
развитии техники и естествознания. В конце позапрошлого и в начале 
прошлого веков исследования в этой области стимулировались главным 
образом потребностями возникшей в то время теплоэнергетики. В годы 
после второй мировой войны развитие авиации, атомной энергетики, 
ракетно-космической техники выдвинуло новые постановки задач 
теплообмена и вместе с тем – новые, более жесткие требования к 
полноте и надежности прогностических возможностей теории. 
 За последние десятилетия сфера интенсивного исследования и 
применения явлений теплообмена чрезвычайно расширилась. Она 
включает как ведущие направления техники (химическая технология, 
металлургия, строительное дело, нефтеразработка, машиностроение, 
агротехника и т.д.), так и основные естественные науки (биология, 
геология, физика атмосферы и океана и другие). Теоретическое 
исследование процессов теплообмена в настоящее время в значительной 
степени базируется на их численном моделировании с использованием 
ЭВМ. Это стало возможным благодаря значительному прогрессу в 
развитии вычислительных методов решения задач для уравнений в 
частных производных и увеличению мощности современных 
вычислительных машин. 
 Нужно особо отметить еще следующие обстоятельства. 
Численное моделирование процессов теплообмена в настоящее время 
приобретает все более значительную роль в связи с тем, что для 
современной науки и техники необходим достоверный прогноз таких 
процессов, экспериментальное изучение которых в лабораторных или 
натурных условиях очень сложно и дорого, а в некоторых случаях 
просто невозможно. Численное моделирование процессов 
теплопереноса все успешнее входит в практику работы различных 
научно-исследовательских, проектно-конструкторских и 
производственных учреждений. 
 Как известно, существуют три механизма передачи тепла: 
теплопроводность, конвекция и излучение. В настоящем пособии мы 
сконцентрируем внимание на основном механизме переноса тепла – 
теплопроводности. 
 Теплопроводностью называется молекулярный перенос теплоты в 
сплошной среде. Этот процесс возникает при неравномерном 
распределении температур. В этом случае теплота передается за счет 
непосредственного соприкосновения частиц, имеющих различную 

температуру, что приводит к обмену энергией между молекулами, 
атомами или свободными электронами. 
 Теплопроводность зависит от агрегатного состояния вещества, его 
состава, чистоты, температуры, давления и других характеристик. Так, в 
большинстве случаев теплопроводность вещества в жидком состоянии 
примерно в 10 раз больше, чем теплопроводность в газообразном 
состоянии. Для твердого тела она значительно выше, чем для жидкости 
около точки плавления (за исключением жидких висмута, олова, 
теллура). 
 На практике часто случается, что теплопроводность внутри тела и 
вблизи его границ различна. Это различие обусловлено как изменением 
условий протекания процессов теплопереноса, так и изменением 
структуры вещества (в результате термообработки, наклепа, старения, 
износа и т.д.). 
 Существенное влияние на теплопроводность могут оказывать 
внешние факторы, например облучение, изменение давления, магнитное 
поле. 
 В полупрозрачных средах теплопроводность сопровождается 
радиационным теплопереносом. Наблюдаемая эффективная 
теплопроводность таких сред есть сумма собственно теплопроводности 
и радиационного теплопереноса. Вклад радиационной составляющей 
комбинированного теплопереноса увеличивается с повышением 
температуры и становится существенным при температурах, 
составляющих несколько сотен градусов Цельсия. 
 Нестационарный перенос тепла теплопроводностью описывается 
следующим уравнением, записанным в декартовой системе координат: 

 Это уравнение (уравнение Фурье – Кирхгофа) устанавливает связь 
между временным и пространственным изменением температуры в 
любой точке тела. Здесь ρ – плотность, c – удельная теплоемкость, λ – 
коэффициент теплопроводности, ( )TtzyxQw ,,,,  – мощность внутренних 
источников тепловыделения. 
 Уравнение (1) описывает множество вариантов развития процесса 
кондуктивного теплопереноса (теплопроводности). Чтобы из 
бесчисленного количества этих вариантов выбрать один и дать его 
полное математическое описание, к соотношению (1) необходимо 
добавить условия однозначности, которые содержат геометрические, 
физические, начальные и граничные условия. 

 Геометрические условия определяют форму и размеры тела, в 
котором протекает изучаемый процесс. Физические условия 
определяют теплофизические характеристики тела λ, ρ, c. Временные 
(начальные) условия содержат распределение температуры в теле в 
начальный момент времени: 

( )zyxfTt ,,  :0 ==  – в общем виде. 
 При равномерном распределении температуры в теле начальное 
условие упрощается: const  :0 0 === TTt . Граничные условия 
определяют особенности протекания процесса на поверхности тела и 
могут быть заданы несколькими способами. 
• Граничные условия первого рода – задается распределение 
температуры на поверхности (или границе) тела для каждого момента 
времени: 

( )tzyxTT w ,,,= , 
где wT  – температура на поверхности тела. Во многих практически 
значимых вариантах const=wT . 
• Граничные условия второго рода – задается значение теплового 
потока для каждой точки поверхности (или границы) тела в любой 
момент времени: 

λ− r , 

где nr  – нормаль к поверхности тела. Наиболее часто используется 
условие const=wq . 
 Такой вариант теплообмена имеет место, например, при 
нагревании различных изделий в высокотемпературных печах. 
• Граничные условия третьего рода – задается взаимосвязь между 
потоком тепла за счет теплопроводности от твердой стенки и тепловым 
потоком из окружающей среды за счет температурного напора (закон 
Ньютона – Рихмана): 

( )eTT

где α – коэффициент теплообмена. Это наиболее широко применяемое 
условие в задачах теплотехники. 
• Граничные условия четвертого рода – для определения теплового 
взаимодействия между элементами, имеющими различные 
теплофизические характеристики, задают условия равенства температур 
и тепловых потоков по обе стороны от границы раздела: 

где ΓΓΓ zyx ,,  – координаты границы раздела сред; 21  ,TT  – температуры 
соприкасающихся сред. Это условие применяется, например, при 
решении задач теплопроводности для многослойных пластин. 
 Дифференциальное уравнение (1) вместе с условиями 
однозначности дает полную математическую формулировку краевой 
задачи теплопроводности. 
 При решении конкретных краевых задач нестационарной 
теплопроводности можно, применяя методы математического 
моделирования, добиться существенного упрощения общей 
математической постановки. Так, если для рассматриваемого процесса: 

то можно вместо уравнения (1) ограничиться одномерным 
нестационарным уравнением кондуктивного теплопереноса 

которое вместе с условиями однозначности дает более простую 
математическую формулировку краевой задачи. Есть много 
практически значимых случаев, когда решение уравнения (2) 
достаточно для полного описания рассматриваемого процесса. 
 В практике теплотехнических расчетов часто возникают 
одномерные задачи с цилиндрической или сферической симметрией. 
Например, цилиндрическая симметрия имеется в задачах об остывании 
длинного цилиндра или при анализе теплового состояния в трубчатых 
каналах. 
 Естественной системой координат в таких задачах является, 
соответственно, цилиндрическая ( )ϕ,r  или сферическая ( )ϕθ,,r . 
Вследствие одномерности все величины не будут зависеть от углов 
ϕθ  , . Тогда параболическое уравнение (2) с переменными 

коэффициентами в соответствующих координатах примет вид: 

где r – радиальная координата, ν – показатель симметрии, равный 0, 1, 2 
соответственно для плоского, цилиндрического и сферического случаев. 

1. МЕТОД КОНЕЧНЫХ РАЗНОСТЕЙ (МКР) 
 
 Сформулированное уравнение (1) с соответствующими краевыми 
условиями (начальными и граничными) будем решать численно, т.е. 
воспользуемся возможностями ЭВМ. Численным решением называется 
решение, полученное в виде таблицы чисел. 
 При решении дифференциального уравнения в частных 
производных наиболее часто используется метод конечных разностей 
(МКР) [1]. Идея МКР решения краевых задач весьма проста и видна уже 
из самого названия: вместо производных в дифференциальном 
уравнении используются их конечноразностные аппроксимации. При 
построении дискретных аппроксимаций краевых дифференциальных 
задач нужно стремиться увязать две, возможно, противоречивые цели: 
хорошее качество аппроксимации и эффективное устойчивое решение 
получающихся при этом алгебраических систем. 
 При использовании МКР для задач теплопроводности твердое 
тело представляют в виде совокупности узлов. Аппроксимируя 
(заменяя) частные производные дифференциального уравнения (1) 
конечными разностями получают систему линейных алгебраических 
уравнений для определения температуры, как локальной 
характеристики в каждом узле сетки. Полученная система является 
незамкнутой, для ее замыканию используют разностное представление 
граничных условий. В результате получают замкнутую систему 
линейных алгебраических уравнений, которую решают численными 
методами с помощью ЭВМ. 
 

2. ЛИНЕЙНЫЕ ЗАДАЧИ ТЕПЛОПРОВОДНОСТИ 
2.1. ОДНОМЕРНОЕ УРАВНЕНИЕ ТЕПЛОПРОВОДНОСТИ 

 
 В качестве примера применения метода конечных разностей 
рассмотрим краевую задачу на основе одномерного уравнения 
теплопроводности. Анализируется теплопередача через плоскую 
бесконечную пластину или изолированный стержень (рис. 1). На одной 
границе пластины поддерживается постоянная температура лT , на 
другой границе – температура пT . Начальная температура равна 0T , 
источники тепловыделения внутри пластины отсутствуют. 

Рис. 1. Геометрия задачи 

 
 При заданных условиях температура будет изменяться только в 
направлениях, перпендикулярных границе пластины. Если ось Ох 
направить, как показано на рис. 1, то температура в направлении Оу и 
Оz может считаться постоянной. Также предположим, что 
теплофизические характеристики не зависят от температуры. В связи с 
этим дифференциальное уравнение (1) преобразуется к виду: 

 Начальные и граничные условия запишутся следующим образом: 

 Для того чтобы дать полное математическое описание 
рассматриваемой задачи, необходимо еще задать физические условия 
однозначности. Если пластина изготовлена из стали, то λ = 46 Вт/(м⋅ºC), 
ρ = 7800 кг/м3, с = 460 Дж/(кг⋅ºC). 
 Эту задачу в полной математической постановке будем решать 
методом конечных разностей на равномерной сетке. Для этого разобьем 
пластину по толщине на N–1 равных промежутков, т.е. построим 
конечно-разностную сетку (рис. 2): 

Рис. 2. Конечно-разностная сетка: 

132 ,, , −Nxxx K  – координаты внутренних узлов; 

Nxx  ,1  – координаты граничных узлов 
 
 Определим значение температуры в i-ом узле в момент времени 

τ⋅== ntt n  как ( ) n
ini TtxT =, . Здесь τ – шаг интегрирования по 

временной координат, n – номер шага по времени. 
 Далее заменим дифференциальные операторы в (3) на их конечно-
разностные аналоги. Будем пользоваться неявной схемой. 

 В результате аппроксимации частных производных 
соответствующими конечными разностями получаем следующую 
систему линейных алгебраических уравнений (СЛАУ): 

 Выбранную схему аппроксимации частных производных можно 
графически представить следующим образом: 

 
Рис. 3. Шаблон неявной четырехточечной разностной схемы 

 Рис. 3 наглядно демонстрирует, что используется 
четырехточечная разностная схема – три точки берутся на новом 
временном слое и одна со старого временного слоя. 
 Сформулированный выше способ аппроксимации производных 
называется неявным потому, что поле температуры на новом временном 
слое представлено неявно, т.е. для его определения необходимо решать 
систему уравнений (5). 
 Полученную систему можно свести к наиболее общему виду: 

 Такие уравнения называют трехточечными разностными 
уравнениями второго порядка. Система (6) имеет трехдиагональную 
структуру. В связи с тем, что рассматривается нестационарная задача, 
систему (6) необходимо решать на каждом шаге по времени. 
 Предположим, что существуют такие наборы чисел 

( )1,1  и −=βα Niii , при которых 

т.е. трехточечное уравнение второго порядка (6) преобразуется в 
двухточечное уравнение первого порядка (7). Уменьшим в связи (7) 
индекс на единицу и полученное выражение 1

 Последнее равенство имеет вид (7) и будет точно с ним совпадать, 
если при всех 1,,3,2 −= Ni K  выполняются соотношения 

 Для определения ii βα  и  по (8) необходимо знать 11  и βα , которые 
находятся из левого граничного условия. 
 Далее по формулам (7) последовательно находятся 

1

1
1 ,,, ++

−

N

n
N TTT K , при условии, что 1+n

NT  найдено из правого граничного 
условия. 

 Таким образом, решение уравнений вида (6) описываемым 
способом, называемым методом прогонки, сводится к вычислениям по 
трем формулам: нахождение так называемых прогоночных 
коэффициентов ii βα  ,  по формулам (8) при 1,2 −= Ni  (прямая 
прогонка) и затем получение неизвестных 1+n

iT  по формуле (7) при 
2,,2,1 K−−= NNi  (обратная прогонка). 

 Для успешного применения метода прогонки нужно, чтобы в 
процессе вычислений не возникло ситуаций с делением на нуль, а при 
больших размерностях систем не должно быть быстрого роста 
погрешностей округлений. 
 Будем называть прогонку корректной, если знаменатели 
прогоночных коэффициентов (8) не обращаются в нуль, и устойчивой, 
если 1<αi  при всех 1,1 −= Ni . 
 В [1] доказана теорема, представляющая достаточные условия 
корректности и устойчивости прогонки уравнений (6): 

1,2    −=∀+> NiCAB iii  и 111 <α⇒<α i ,  (9) 
которые во многих приложениях метода выполняются автоматически. 
 Возвращаясь к системе (5), определим прогоночные 
коэффициенты и воссоздадим полный алгоритм решения полученной 
системы. 
 Поскольку при 0=x  лTT = , 
то  

л
nn ТTT =β+⋅α= ++

1

21

а при 
Lx =   пTT = ,  

п
n

N ТT =+1 . 
 Прогоночные коэффициенты вычисляются по формулам (8). 
 Таким образом, разностные соотношения, аппроксимирующие 
дифференциальную задачу (3), (4), имеют следующий вид: 

 Аппроксимация дифференциальной задачи (3), (4) конечно-
разностной (10), (11) выполнена с первым порядком точности по 
времени t и вторым по пространственной координате h. При этом 
неявная разностная схема является абсолютно устойчивой, т.е. можно 
проводить интегрирование краевой задачи (3), (4) с любым разностным 
шагом по времени. Шаг по времени выбирается таким образом, чтобы 
весь интервал времени разбивался хотя бы на 10 шагов (желательно 
больше). 
 

Блок-схема к рассматриваемой задаче имеет вид: 

Ниже приведен листинг программы для решения рассматриваемой 
задачи (на языке программирования Pascal) 

 
uses crt; 
const mf=500; 
type 
     vector=array[1..mf] of real; 
var {раздел описания переменных, которые мы будем использовать в 
программе} 
 i, j, N                                     : integer; 
 T, alfa, beta                           : vector; 
 ai, bi, ci, fi                             : real; 
 lamda, ro, c, h, tau                 : real; 
 Tl, T0, Tr, L, t_end, time      : real; 
 f, g                                         : text; 
begin 
 clrscr; 
{с клавиатуры вводим все необходимые входные параметры} 
 Writeln('Введите количество пространственных узлов, N'); 
 Readln(N); 
 Writeln('Введите окончание по времени, t_end'); 
 Readln(t_end); 
 Writeln('Введите толщину пластины, L'); 
 Readln(L); 
 Writeln('Введите коэффициент теплопроводности материала пластины, 
lamda'); 
 Readln(lamda); 
 Writeln('Введите плотность материала пластины, ro'); 
 Readln(ro); 
 Writeln('Введите теплоемкость материала пластины, c'); 
 Readln(c); 
 Writeln('Введите начальную температуру, T0'); 
 Readln(T0); 
 Writeln('Введите температуру на границе х=0, Tl'); 
 Readln(Tl); 
 Writeln('Введите температуру на границе х=L, Tr'); 
 Readln(Tr); 
{определяем расчетный шаг сетки по пространственной координате} 
 h:=L/(N-1); 
{определяем расчетный шаг сетки по времени} 
 tau:=t_end/100.0; 

{определяем поле температуры в начальный момент времени} 
 for i:= 1 to N do 
  T[i]:=T0; 
{проводим интегрирование нестационарного уравнения 
теплопроводности} 
 time:=0; 
 while time<t_end do {используем цикл с предусловием} 
  begin 
   time:=time+tau; 
{определяем начальные прогоночные коэффициенты на основе левого 
граничного условия} 
   alfa[1]:=0.0; 
   beta[1]:=Tl; 
{цикл с параметром для определения прогоночных коэффициентов по 
формуле (8)} 
   for i:= 2 to N-1 do 
    begin 
{ai, bi, ci, fi – коэффициенты канонического представления СЛАУ с 
трехдиагональной матрицей} 
     ai:=lamda/sqr(h); 
     bi:=2.0*lamda/sqr(h)+ro*c/tau; 
     ci:=lamda/sqr(h); 
     fi:=-ro*c*T[i]/tau; 
{alfa[i], beta[i] – прогоночные коэффициенты} 
     alfa[i]:=ai/(bi-ci*alfa[i-1]); 
     beta[i]:=(ci*beta[i-1]-fi)/(bi-ci*alfa[i-1]); 
    end; 
{определяем значение температуры на правой границе} 
   T[N]:=Tr; 
{используя соотношение (7) определяем неизвестное поле 
температуры} 
   for i:= N-1 downto 1 do 
    T[i]:=alfa[i]*T[i+1]+beta[i]; 
  end; {цикл с предусловием окончен} 
{выводим результаты в файл} 
 Assign(f,'res.txt'); 
 Rewrite(f); 
 Writeln(f,'Толщина пластины L = ',L:6:4); 
 Writeln(f,'Число узлов по координате N = ',N); 
 Writeln(f,'Коэффициент теплопроводности материала пластины lamda = 
',lamda:6:4); 

 Writeln(f,'Плотность материала пластины ro = ',ro:6:4); 
 Writeln(f,'Теплоемкость материала пластины с = ',c:6:4); 
 Writeln(f,'Начальная температура  T0 = ',T0:6:4); 
 Writeln(f,'Температура на границе x = 0, Tl = ',Tl:6:4); 
 Writeln(f,'Температура на границе x = L, Tr = ',Tr:6:4); 
 Writeln(f,'Результат получен с шагом по координате h = ',h:6:4); 
 Writeln(f,'Результат получен с шагом по времени tau = ',tau:6:4); 
 Writeln(f,'Температурное поле в момент времени t = ',t_end:6:4); 
 close(f); 
 Assign(g,'tempr.txt'); 
 Rewrite(g); 
 for i:=1 to N do 
  writeln(g,'  ',h*(i-1):6:3,'   ',T[i]:8:5); 
 close(g); 
end. 
 
 Результаты расчетов по приведенной программе при L = 0,1 м,  
λ = 46 Вт/(м⋅ºC), ρ = 7800 кг/м3, с = 460 Дж/(кг⋅ºC), Т0=20 °С, Тл = 300 °С, 
Тп = 100 °С через 60 секунд процесса нагрева приведены на рис. 4. 
 

 
Рис. 4. Распределение температуры по толщине пластины в момент 

времени t = 60 с 
 
 Как отмечалось выше, рассмотренная расчетная схема является 
неявной, т.е. для определения поля температуры приходится решать 
систему линейных алгебраических уравнений. Но помимо 

предложенной схемы существует также и явная схема. В такой схеме 
явно определяется поле температуры и не нужно решать систему 
уравнений для определения прогоночных коэффициентов iα  и iβ . 
Рассмотрим ту же задачу, но уже с использованием явной схемы. 
 Отличие явной схемы от неявной заключается в аппроксимации 
диффузионного слагаемого, а именно, во временном слое на котором 
рассматривается неизвестное поле температуры: 

 Таким образом, в результате аппроксимации частных 
производных соответствующими конечными разностями получаем 
следующее соотношения для определения поля температуры: 

 Графически явную разностную схему можно представить 
следующим образом: 
 

 
Рис. 5. Шаблон явной четырехточечной разностной схемы 

 
 Из шаблона (рис. 5) видно, что для определения неизвестного 
поля температуры никакой системы уравнений для iα  и iβ  решать не 
требуется. 

и аналогичные разностные аналоги краевых условий: 

 Таким образом, мы получили простую систему линейных 
алгебраических уравнений для нахождения распределения температуры 
в пластине в различные моменты времени. Аппроксимация 
дифференциальной задачи (3), (4) конечно-разностной (12), (13) 
выполнена также с первым порядком по времени t и вторым по 
пространственной координате h. Но чтобы решение конечно-разностной 
задачи (12), (13) сходилось к решению дифференциальной задачи, 
достаточно выполнение следующего условия (условие устойчивости 
разностной схемы): 

λ⋅
⋅⋅ρ

<τ
2

2hc . 

Из этого условия определяется шаг интегрирования по временной 
координате. 
 Таким образом, явная разностная схема является условно 
устойчивой и требует специальных мероприятий по оценке 
возможности ее использования. 
 

Блок-схема к рассматриваемой задаче (явная схема) имеет вид: 

Ниже приведен листинг программы (явная схема) для решения 
рассматриваемой задачи (на языке программирования Pascal) 

 
uses crt; 
const mf=500; 
type 
     vector=array[0..mf] of real; 
var {раздел описания переменных, которые мы будем использовать в 
программе} 
 i, j, N                                    : integer; 
 T, TT                                    : vector; 
 a, lamda, ro, c, h, tau            : real; 
 Tl, T0, Tr, L, t_end, time     : real; 
 f, g                                        : text; 
begin 
 clrscr; 
{с клавиатуры вводим все необходимые входные параметры} 
 Writeln('Введите количество узлов по пространственной координате, 
N'); 
 Readln(N); 
 Writeln('Введите окончание по времени, t_end'); 
 Readln(t_end); 
 Writeln('Введите толщину пластины, L'); 
 Readln(L); 
 Writeln('Введите коэффициент теплопроводности материала пластины, 
lamda'); 
 Readln(lamda); 
 Writeln('Введите плотность материала пластины, ro'); 
 Readln(ro); 
 Writeln('Введите теплоемкость материала пластины, c'); 
 Readln(c); 
 Writeln('Введите начальную температуру, T0'); 
 Readln(T0); 
 Writeln('Введите температуру на границе х=0, Tl'); 
 Readln(Tl); 
 Writeln('Введите температуру на границе х=L, Tr'); 
 Readln(Tr); 
{определяем коэффициент температуропроводности} 
 a:=lamda/(ro*c); 
{определяем расчетный шаг сетки по пространственной координате} 
 h:=L/(N-1); 

{определяем расчетный шаг сетки по времени, руководствуясь 
условием устойчивости} 
 tau:=0.25*sqr(h)/a; 
{определяем поле температуры в начальный момент времени} 
 for i:= 2 to N-1 do 
  T[i]:=T0; 
{определяем значения температуры на границе} 
 T[1]:=Tl; 
 T[N]:=Tr; 
{проводим интегрирование нестационарного уравнения 
теплопроводности} 
 time:=0; 
 while time<t_end do {используем цикл с предусловием} 
  begin 
{увеличиваем переменную времени на шаг} 
   time:=time+tau; 
{запоминаем поле температуры на предыдущем слое по времени} 
   for i:= 1 to N do  
    TT[i]:=T[i]; 
{определяем неизвестное поле температуры по соотношениям (12)} 
   for i:= 2 to N-1 do 
    T[i]:=TT[i]+a*tau/sqr(h)*(TT[i+1]-2.0*TT[i]+TT[i-1]); 
  end; 
{выводим результат в файл} 
 Assign(f,'res.txt'); 
 Rewrite(f); 
 Writeln(f,'Толщина пластины L = ',L:6:4); 
 Writeln(f,'Число узлов по пространственной координате  N = ',N); 
 Writeln(f,'Коэффициент теплопроводности материала пластины lamda = 
',lamda:6:4); 
 Writeln(f,'Плотность материала пластины ro = ',ro:6:4); 
 Writeln(f,'Теплоемкость материала пластины с = ',c:6:4); 
 Writeln(f,'Начальная температура  T0 = ',T0:6:4); 
 Writeln(f,'Температура на границе x = 0, Tl = ',Tl:6:4); 
 Writeln(f,'Температура на границе x = L, Tr = ',Tr:6:4); 
 Writeln(f,'Результат получен с шагом по пространственной координате 
x, h = ',h:6:4); 
 Writeln(f,'Результат получен с шагом по времени tau = ',tau:6:4); 
 Writeln(f,'Температурное поле в момент времени t = ',t_end:6:4); 
 close(f); 
 Assign(g,'tempr.txt'); 

 Rewrite(g); 
 for i:=1 to N do 
  writeln(g,'  ',h*(i-1):6:3,'   ',T[i]:8:5); 
 close(g); 
end. 
 
 Результаты расчетов по приведенной программе при L = 0,1 м,  
λ = 46 Вт/(м⋅ºC), ρ = 7800 кг/м3, с = 460 Дж/(кг⋅ºC), Т0=20 °С, Т1 = 300 °С, 
Т2 = 100 °С через 60 секунд процесса нагрева приведены на рис. 6. 
 

 
Рис. 6. Распределение температуры по толщине пластины в момент 

времени t = 60 с (с использованием явной разностной схемы) 
 
 Анализируя рисунки 4 и 6, видим, что полученные поля 
температур идентичны. Покажем, в чем же заключается недостаток 

явных схем. При использовании данной схемы 
λ⋅
⋅⋅ρ

<τ
2

2hc . Рассмотрим 

случай 
λ
⋅⋅ρ

=τ
2hc  как для явной, так и для неявной схемы. С помощью 

программа, построенной на основе явной схемы, рассчитан только 
случай N = 50, t = 20. При N > 50 и t > 20 происходит ошибка (в случае 
среды Turbo Pascal – переполнение операции с плавающей точкой). 

а 

 
б 

Рис. 7. Распределение температуры по толщине пластины в момент 
времени t = 20 с: а – явная схема, б – неявная схема 

 
 Рис. 7,а наглядно демонстрирует появление осцилляций при 
использовании явной схемы, что характеризует её неустойчивость. 
Неявная же схема напротив дает корректный результат. 
 Поэтому дальнейшее изложение материала будем проводить с 
использованием неявной разностной схемы. 
 Рассмотрим одномерное уравнение теплопроводности в 
цилиндрических координатах. 
 Определим температурное поле в бесконечном цилиндре радиуса 

м 1.0=R  через 5, 10, 30 и 60 секунд. Материал цилиндра – кирпич       
(λ = 0,7 Вт/(м⋅ºC), ρ = 1500 кг/м3, с = 750 Дж/(кг⋅ºC)). Начальная 
температура С200

0 =T . На границе Rr =  поддерживается постоянная 
температура C 50 0=hT . 
 
 
 
 
 
 
 
 
 
 
 
 



 25

 
Рис. 8. Геометрия задачи 

 

 Математическая постановка задачи будет иметь вид: 

 Начальные и граничные условия запишутся следующим образом: 

 Для решения сформулированной краевой задачи применим метод 
конечных разностей на основе неявной четырехточечной схемы. 
Сначала введем равномерную пространственную сетку: 

 Аналогично вводится временная сетка: 

 Заменим дифференциальные операторы в уравнении (14) на их 
конечно-разностные аналоги: 

 Таким образом, в результате аппроксимации частных 
производных соответствующими конечными разностями получаем 
следующую систему линейных алгебраических уравнений: 

 Полученную систему можно свести к наиболее общему виду: 

 Прогоночные коэффициенты находятся по формулам (8). Далее 
неизвестное поле температуры определяется по выражению (7). 
 Воспользуемся левым граничным условием для определения 
начальных прогоночных коэффициентов 11  и βα  из соотношения 

1211 β+⋅α= TT . На левой границе стоит условие симметрии: 

 А на правой границе температура известна .hRr TT =
=

Блок-схема к рассматриваемой задаче имеет вид: 

Ниже приведен листинг программы для решения рассматриваемой 
задачи (на языке программирования Pascal) 

 
uses crt; 
const mf=500; 
type 
     vector=array[1..mf] of real; 
var {раздел описания переменных, которые мы будем использовать в 
программе} 
 i, j, N                                : integer; 
 T, alfa, beta                      : vector; 
 ai, bi, ci, fi                        : real; 
 lamda, ro, c, h, tau            : real; 
 T0, Th, R, t_end, time      : real; 
 f, g                                    : text; 
begin 
 clrscr; 
{с клавиатуры вводим все необходимые входные параметры} 
 Writeln('Введите количество пространственных узлов, N'); 
 Readln(N); 
 Writeln('Введите окончание по времени, t_end'); 
 Readln(t_end); 
 Writeln('Введите радиус цилиндра, R'); 
 Readln(R); 
 Writeln('Введите коэффициент теплопроводности материала пластины, 
lamda'); 
 Readln(lamda); 
 Writeln('Введите плотность материала пластины, ro'); 
 Readln(ro); 
 Writeln('Введите теплоемкость материала пластины, c'); 
 Readln(c); 
 Writeln('Введите начальную температуру, T0'); 
 Readln(T0); 
 Writeln('Введите температуру на границе r=R, Th'); 
 Readln(Th); 
{определяем расчетный шаг сетки по пространственной координате} 
 h:=R/(N-1); 
{определяем расчетный шаг сетки по времени} 
 tau:=t_end/100.0; 
{определяем поле температуры в начальный момент времени} 
 for i:= 1 to N do 

  T[i]:=T0; 
{проводим интегрирование нестационарного уравнения 
теплопроводности} 
 time:=0; 
 while time<t_end do {используем цикл с предусловием} 
  begin 
{увеличиваем переменную времени на шаг τ} 
   time:=time+tau; 
{определяем начальные прогоночные коэффициенты на основе левого 
граничного условия} 
   alfa[1]:=1.0; 
   beta[1]:=0.0; 
{цикл с параметром для определения прогоночных коэффициентов по 
формуле (8)} 
   for i:= 2 to N-1 do 
    begin 
{ai, bi, ci, fi – коэффициенты канонического представления СЛАУ с 
трехдиагональной матрицей} 
     ai:=0.5*lamda*(2*i-1)/(sqr(h)*(i-1)); 
     ci:=0.5*lamda*(2*i-3)/(sqr(h)*(i-1)); 
     bi:=ai+ci+ro*c/tau; 
     fi:=-ro*c*T[i]/tau; 
{alfa[i], beta[i] – прогоночные коэффициенты} 
     alfa[i]:=ai/(bi-ci*alfa[i-1]); 
     beta[i]:=(ci*beta[i-1]-fi)/(bi-ci*alfa[i-1]); 
    end; 
{определяем значение температуры на правой границе} 
   T[N]:=Th; 
{используя соотношение (7) определяем неизвестное поле 
температуры} 
   for i:= N-1 downto 1 do 
    T[i]:=alfa[i]*T[i+1]+beta[i]; 
  end; 
{выводим результат в файл} 
 Assign(f,'res.txt'); 
 Rewrite(f); 
 Writeln(f,'Радиус цилиндра R = ',R:6:4); 
 Writeln(f,'Число узлов по координате N = ',N); 
 Writeln(f,'Коэффициент теплопроводности материала пластины lamda = 
',lamda:6:4); 
 Writeln(f,'Плотность материала пластины ro = ',ro:6:4); 

 Writeln(f,'Теплоемкость материала пластины с = ',c:6:4); 
 Writeln(f,'Начальная температура  T0 = ',T0:6:4); 
 Writeln(f,'Температура на границе r = R, Th = ',Th:6:4); 
 Writeln(f,'Результат получен с шагом по координате h = ',h:6:4); 
 Writeln(f,'Результат получен с шагом по времени tau = ',tau:6:4); 
 Writeln(f,'Температурное поле в момент времени t = ',t_end:6:4); 
 close(f); 
 Assign(g,'tempr.txt'); 
 Rewrite(g); 
 for i:=1 to N do 
  writeln(g,'  ',h*(i-1):6:3,'   ',T[i]:8:5); 
 close(g); 
end. 
 
 В результате были получены следующие распределения 
температуры 
 

 
Рис. 9. Распределения температуры по радиусу цилиндра в различные 

моменты времени 
 

2.2. КОНЕЧНО-РАЗНОСТНАЯ АППРОКСИМАЦИЯ 
ГРАНИЧНЫХ УСЛОВИЙ ВТОРОГО И ТРЕТЬЕГО РОДА 

 
 Если для задачи, изложенной в пункте 2.1, в качестве граничных 
условий на внешних поверхностях использовать граничные условия 

второго рода (заданы тепловые потоки q1 и q2, рис. 10), тогда 
математическая формулировка граничных условий будет иметь 
следующий вид: 

Рис. 10. Геометрия задачи (граничные условия второго рода) 

 
 Проанализируем соотношения (16): 

материала; нагрев происходит 0 границе на 01 =⇒> xq  
материала; охлаждение происходит 0 границе на 01 =⇒< xq  
материала; охлаждение происходит  границе на 02 Lxq =⇒>  

материала. нагрев происходит  границе на 02 Lxq =⇒<  
 Проведем дискретизацию граничных условий II рода с 
погрешностью )(hO . Погрешность аппроксимации вида )(hO  означает, 
что различия между точными значениями и полученными 
(приближенными) будут одного порядка с шагом по пространству h. 
Поскольку обычно 1<h , то погрешность будет тем меньше, чем выше 
порядок аппроксимации. Например, погрешность аппроксимации вида 

)( 2hO  дает результаты более точные, т.к. 12 << hh . 
 Поскольку мы будем использовать неявную разностную схему, то 
левое граничное условие необходимо для определения первых 
прогоночных коэффициентов 11  и βα  из соотношения 1211 β+⋅α= TT . 
Итак,  

 Отсюда следует, что 

 А правое граничное условие используют для определения 
температуры NT . 
Итак,  

 Проведем дискретизацию граничных условий II рода с 
погрешностью )( 2hO . Предположим, что на границе выполняется 
уравнение теплопроводности (3): 

где а – коэффициент температуропроводности материала. 
 Разложим функцию )(xТ  в ряд Тейлора в окрестности точки 0=x  
до членов второго порядка относительно h: 

Таким образом, 

 Определим NT  используя правое граничное условие. 

 Граничные условия третьего рода для задачи из пункта 2.1 (если 
температуры окружающей среды e1T  и e2T  и коэффициенты 
теплоотдачи 1κ  и 2κ ) можно сформулировать следующим образом: 

 Проведем дискретизацию граничных условий III рода с 
погрешностью )(hO . 
 Определим первые прогоночные коэффициенты 11  и βα  из 
соотношения 1211 β+⋅α= TT . 

 А правое граничное условие используют для определения 
температуры NT . 
Итак, 

 Проведем дискретизацию граничных условий III рода с 
погрешностью )( 2hO . Предположим, что на границе выполняется 
уравнение теплопроводности (19). Далее по аналогии с аппроксимацией 
граничного условия II рода получим: 

 Использование же правого граничного условия дает следующее 
соотношение: 

2.3. ПРИМЕРЫ КРАЕВЫХ ЗАДАЧ С РАЗЛИЧНЫМИ 

ГРАНИЧНЫМИ УСЛОВИЯМИ 
 

1. Определим температурное поле в медной пластинке через 5, 10, 
30 и 60 секунд. Толщина пластины м3.0=L . Начальная температура 

С200
0 =T . Медь имеет следующие теплофизические характеристики     

λ = 384 Вт/(м⋅ºC), ρ = 8800 кг/м3, с = 381 Дж/(кг⋅ºC). На границе 0=x  
приложен тепловой поток 27 мВт 10=q , а граница Lx =  подвержена 
воздействию внешней среды ( ( )СмВт 100 02 ⋅=κ , C 300 0e =T ). 
 Математическая постановка задачи будет иметь вид: 

Lx

t

 Начальные и граничные условия запишутся следующим образом: 

Блок-схема к рассматриваемой задаче имеет вид: 

Ниже приведен листинг программы для решения рассматриваемой 
задачи (на языке программирования Pascal) 

 
uses crt; 
const mf=500; 
type 
     vector=array[1..mf] of real; 
var {раздел описания переменных, которые мы будем использовать в 
программе} 
 i, j, N                                   : integer; 
 T, alfa, beta                         : vector; 
 ai, bi, ci, fi                           : real; 
 a, lamda, ro, c, h, tau          : real; 
 q, kapa, Te                          : real; 
 T0, L, t_end, time               : real; 
 f, g                                       : text; 
begin 
 clrscr; 
{с клавиатуры вводим все необходимые входные параметры} 
 Writeln('Введите количество пространственных узлов, N'); 
 Readln(N); 
 Writeln('Введите окончание по времени, t_end'); 
 Readln(t_end); 
 Writeln('Введите толщину пластины, L'); 
 Readln(L); 
 Writeln('Введите коэффициент теплопроводности материала пластины, 
lamda'); 
 Readln(lamda); 
 Writeln('Введите плотность материала пластины, ro'); 
 Readln(ro); 
 Writeln('Введите теплоемкость материала пластины, c'); 
 Readln(c); 
 Writeln('Введите плотность теплового потока, q'); 
 Readln(q); 
 Writeln('Введите коэффициент теплообмена, kapa'); 
 Readln(kapa); 
 Writeln('Введите температуру внешней среды, Te'); 
 Readln(Te); 
 Writeln('Введите начальную температуру, T0'); 
 Readln(T0); 

{определяем расчетный шаг сетки по пространственной координате} 
 h:=L/(N-1); 
{определяем коэффициент температуропроводности} 
 a:=lamda/(ro*c); 
{определяем расчетный шаг сетки по времени} 
 tau:=t_end/100.0; 
{определяем поле температуры в начальный момент времени} 
 for i:= 1 to N do 
  T[i]:=T0; 
{проводим интегрирование нестационарного уравнения 
теплопроводности} 
 time:=0; 
 while time<t_end do {используем цикл с предусловием} 
  begin 
{увеличиваем переменную времени на шаг τ} 
   time:=time+tau; 
{определяем начальные прогоночные коэффициенты на основе левого 
граничного условия, используя соотношения (20)} 
   alfa[1]:=2.0*a*tau/(2.0*a*tau+sqr(h)); 
   beta[1]:=(sqr(h)*T[1]+2.0*a*tau*h*q/lamda)/(2.0*a*tau+sqr(h)); 
{цикл с параметром для определения прогоночных коэффициентов по 
формуле (8)} 
   for i:= 2 to N-1 do 
    begin 
{ai, bi, ci, fi – коэффициенты канонического представления СЛАУ с 
трехдиагональной матрицей} 
     ai:=lamda/sqr(h); 
     bi:=2.0*lamda/sqr(h)+ro*c/tau; 
     ci:=lamda/sqr(h); 
     fi:=-ro*c*T[i]/tau; 
{alfa[i], beta[i] – прогоночные коэффициенты} 
     alfa[i]:=ai/(bi-ci*alfa[i-1]); 
     beta[i]:=(ci*beta[i-1]-fi)/(bi-ci*alfa[i-1]); 
    end; 
{определяем значение температуры на правой границе, используя 
соотношение (25)} 
   T[N]:=(lamda*sqr(h)*T[N]+2.0*a*tau*(lamda*beta[N-1]+kapa*h*Te)) 
              /(lamda*sqr(h)+2.0*a*tau*(lamda*(1-alfa[N-1])+kapa*h)); 
{используя соотношение (7) определяем неизвестное поле 
температуры} 
   for i:= N-1 downto 1 do 

    T[i]:=alfa[i]*T[i+1]+beta[i]; 
  end; {цикл с предусловием окончен} 
{выводим результат в файл} 
 Assign(f,'res.txt'); 
 Rewrite(f); 
 Writeln(f,'Толщина пластины L = ',L:6:4); 
 Writeln(f,'Число узлов по координате N = ',N); 
 Writeln(f,'Коэффициент теплопроводности материала пластины lamda = 
',lamda:6:4); 
 Writeln(f,'Плотность материала пластины ro = ',ro:6:4); 
 Writeln(f,'Теплоемкость материала пластины с = ',c:6:4); 
 Writeln(f,'Начальная температура  T0 = ',T0:6:4); 
 Writeln(f,'Плотность теплового потока q = ',q:6:4); 
 Writeln(f,'Коэффициент теплообмена kapa = ',kapa:6:4); 
 Writeln(f,'Температура внешней среды Te = ',Te:6:4); 
 Writeln(f,'Результат получен с шагом по координате h = ',h:6:4); 
 Writeln(f,'Результат получен с шагом по времени tau = ',tau:6:4); 
 Writeln(f,'Температурное поле в момент времени t = ',t_end:6:4); 
 close(f); 
 Assign(g,'tempr.txt'); 
 Rewrite(g); 
 for i:=1 to N do 
  writeln(g,'  ',h*(i-1):10:8,'   ',T[i]:8:5); 
 close(g); 
end. 

 В результате получены следующие распределения температуры 
 

 
Рис. 11. Распределения температуры по толщине пластины в различные 

моменты времени 
 

2. Определим температурное поле в медной пластине через 30, 180 и 
600 секунд. Толщина пластины м3.0=L . Начальная температура 

С500
0 =T . На границе 0=x  и Lx =  пластина контактирует с внешней 

средой ( ( )СмВт 1000 02
1 ⋅=κ , C30 0e1 −=T  и ( )СмВт 500 02

2 ⋅=κ , 
C 10 0e1 =T ). 

 Математическая постановка задачи будет иметь вид: 

 Начальные и граничные условия запишутся следующим образом: 

Блок-схема к рассматриваемой задаче имеет вид: 

Ниже приведен листинг программы для решения рассматриваемой 
задачи (на языке программирования Pascal) 

 
uses crt; 
const mf=500; 
type 
     vector=array[1..mf] of real; 
var {раздел описания переменных, которые мы будем использовать в 
программе} 
 i, j, N                                  : integer; 
 T, alfa, beta                        : vector; 
 ai, bi, ci, fi                          : real; 
 a, lamda, ro, c, h, tau          : real; 
 kapa1, kapa2, Te1, Te2      : real; 
 T0, L, t_end, time               : real; 
 f, g                                      : text; 
begin 
 clrscr; 
{с клавиатуры вводим все необходимые входные параметры} 
 Writeln('Введите количество пространственных узлов, N'); 
 Readln(N); 
 Writeln('Введите окончание по времени, t_end'); 
 Readln(t_end); 
 Writeln('Введите толщину пластины, L'); 
 Readln(L); 
 Writeln('Введите коэффициент теплопроводности материала пластины, 
lamda'); 
 Readln(lamda); 
 Writeln('Введите плотность материала пластины, ro'); 
 Readln(ro); 
 Writeln('Введите теплоемкость материала пластины, c'); 
 Readln(c); 
 Writeln('Введите коэффициент теплообмена на границе х = 0, kapa1'); 
 Readln(kapa1); 
 Writeln('Введите коэффициент теплообмена на границе х = L, kapa2'); 
 Readln(kapa2); 
 Writeln('Введите температуру внешней среды относительно границы х 
= 0, Te1'); 
 Readln(Te1); 
 Writeln('Введите температуру внешней среды относительно границы х 
= L, Te2'); 

 Readln(Te2); 
 Writeln('Введите начальную температуру, T0'); 
 Readln(T0); 
{определяем расчетный шаг сетки по пространственной координате} 
 h:=L/(N-1); 
{определяем коэффициент температуропроводности} 
 a:=lamda/(ro*c); 
{определяем расчетный шаг сетки по времени} 
 tau:=t_end/100.0; 
{определяем поле температуры в начальный момент времени} 
 for i:= 1 to N do 
  T[i]:=T0; 
{проводим интегрирование нестационарного уравнения 
теплопроводности} 
 time:=0; 
 while time<t_end do {используем цикл с предусловием} 
  begin 
{увеличиваем переменную времени на шаг τ} 
   time:=time+tau; 
{определяем начальные прогоночные коэффициенты на основе левого 
граничного условия, используя соотношения (24)} 
   alfa[1]:=2.0*a*tau*lamda/(2.0*a*tau*(lamda+kapa1*h)+lamda*sqr(h)); 
   beta[1]:=(lamda*sqr(h)*T[1]+2.0*a*tau*kapa1*h*Te1) 
                 /(2.0*a*tau*(lamda+kapa1*h)+lamda*sqr(h)); 
{цикл с параметром для определения прогоночных коэффициентов по 
формуле (8)} 
   for i:= 2 to N-1 do 
    begin 
{ai, bi, ci, fi – коэффициенты канонического представления СЛАУ с 
трехдиагональной матрицей} 
     ai:=lamda/sqr(h); 
     bi:=2.0*lamda/sqr(h)+ro*c/tau; 
     ci:=lamda/sqr(h); 
     fi:=-ro*c*T[i]/tau; 
{alfa[i], beta[i] – прогоночные коэффициенты} 
     alfa[i]:=ai/(bi-ci*alfa[i-1]); 
     beta[i]:=(ci*beta[i-1]-fi)/(bi-ci*alfa[i-1]); 
    end; 

{определяем значение температуры на правой границе, используя 
соотношение (25)} 
   T[N]:=(lamda*sqr(h)*T[N]+2.0*a*tau*(lamda*beta[N-1]+kapa2*h*Te2)) 
              /(lamda*sqr(h)+2.0*a*tau*(lamda*(1-alfa[N-1])+kapa2*h)); 
{используя соотношение (7) определяем неизвестное поле 
температуры} 
   for i:= N-1 downto 1 do 
    T[i]:=alfa[i]*T[i+1]+beta[i]; 
  end; {цикл с предусловием окончен} 
{выводим результат в файл} 
 Assign(f,'res.txt'); 
 Rewrite(f); 
 Writeln(f,'Толщина пластины L = ',L:6:4); 
 Writeln(f,'Число узлов по координате N = ',N); 
 Writeln(f,'Коэффициент теплопроводности материала пластины lamda = 
',lamda:6:4); 
 Writeln(f,'Плотность материала пластины ro = ',ro:6:4); 
 Writeln(f,'Теплоемкость материала пластины с = ',c:6:4); 
 Writeln(f,'Начальная температура  T0 = ',T0:6:4); 
 Writeln(f,'Коэффициент теплообмена kapa1 = ',kapa1:6:4); 
 Writeln(f,'Коэффициент теплообмена kapa2 = ',kapa2:6:4); 
 Writeln(f,'Температура внешней среды Te1 = ',Te1:6:4); 
 Writeln(f,'Температура внешней среды Te2 = ',Te2:6:4); 
 Writeln(f,'Результат получен с шагом по координате h = ',h:6:4); 
 Writeln(f,'Результат получен с шагом по времени tau = ',tau:6:4); 
 Writeln(f,'Температурное поле в момент времени t = ',t_end:6:4); 
 close(f); 
 Assign(g,'tempr.txt'); 
 Rewrite(g); 
 for i:=1 to N do 
  writeln(g,'  ',h*(i-1):10:8,'   ',T[i]:8:5); 
 close(g); 
end. 

 Решение задачи имеет вид: 
 

 
Рис. 12. Распределения температуры по толщине пластины в различные 

моменты времени 
 

2.4. ДВУХСЛОЙНАЯ ПЛАСТИНА 
 
 Во многих важных практических приложениях проводится оценка 
температурных полей в многослойных деталях конструкций. 
 Проанализируем процесс теплопереноса в теле, представляющем 
собой совокупность двух пластин с различными теплофизическими 
характеристиками (рис. 13). 
 

 
Рис. 13. Геометрия задачи 

 Математическая постановка задачи будет иметь вид: 

где 1 соответствует левой пластине (1 на рис. 13), 
2 соответствует левой пластине (2 на рис. 13). 
 Начальные и граничные условия можно записать следующим 
образом: 

 На границах 0=x  и Lx =  рассматриваются граничные условия 
первого рода для простоты дальнейшего изложения. Задание на этих 
границах условий II или III рода было подробно изложено в пункте 2.2. 
Принципиальным моментом в настоящем параграфе является 
исследование граничных условий IV рода в точке контакта двух 
пластин. 
 Решение данной задачи проводится также численно на основе 
неявной разностной схемы. Граничное условие IV рода используется 
для определения прогоночных коэффициентов в точке *x . 
 Алгоритм решения сформулированной краевой задачи можно 
представить следующим образом. 
 Сначала проводим аппроксимацию дифференциального 
уравнения конечными разностями, получаем систему линейных 
алгебраических уравнений вида (10), которую решаем методом 
прогонки. При нахождении прогоночных коэффициентов в области 

*0 xx <≤  используем характеристики среды 1, а при Lxx ≤<*  – среды 
2. В точке же *xx =  необходимо использовать граничное условие IV 
рода. 

 Выведем прогоночные коэффициенты в точке контакта двух сред. 
 

 
Рис. 14. Шаблон разностной сетки 

 Рассмотрим аппроксимацию первого порядка относительно шага 
по пространственной координате. При этом, принимая во внимание то, 
что при *ii <  1TT = , а при *ii >  2TT =  исключим в записи индексы, 
характеризующие среду. Получим следующие соотношения: 

 Введем обозначение *** ,2,1 iii TTT ≡= . 
 Используя прогоночное соотношение 111 **** −−− β+⋅α= iiii TT  из 
второго равенства условий (26) получим: 

 В случае же аппроксимации второго порядка относительно шага 
по пространственной координате получим 

 Итак, сначала находят прогоночные коэффициенты для первой 
среды, на границе *i  используют соотношения (27) или (28), а далее 
определяют прогоночные коэффициенты для второй среды. 
 В качестве примера определим температурное поле в составной 
пластинке (рис. 13) через 30, 180 и 600 секунд. Толщина пластины 

м 3.0=L . Будем полагать, что толщины составных частей одинаковые. 
Начальная температура С100

0 =T . Одна часть пластины (1) – сталь      
(λ = 46 Вт/(м⋅ºC), ρ = 7800 кг/м3, с = 460 Дж/(кг⋅ºC)), а другая часть      
(2) – медь (λ = 384 Вт/(м⋅ºC), ρ = 8800 кг/м3, с = 381 Дж/(кг⋅ºC)). На 
границе 0=x  поддерживается постоянная температура C 100 0=T , а на 
границе Lx =  C 50 0=T . 

Блок-схема к рассматриваемой задаче имеет вид: 

Ниже приведен листинг программы для решения рассматриваемой 
задачи (на языке программирования Pascal) 

 
uses crt; 
const mf=500; 
type 
     vector=array[1..mf] of real; 
var {раздел описания переменных, которые мы будем использовать в 
программе} 
 i, j, N, N1, N2            : integer; 
 T, alfa, beta                : vector; 
 ai, bi, ci, fi                  : real; 
 a1, lamda1, ro1, c1     : real; 
 a2, lamda2, ro2, c2     : real; 
 h, tau, t_end, time       : real; 
 T0, Tl, Tr, L               : real; 
 f, g                              : text; 
begin 
 clrscr; 
{с клавиатуры вводим все необходимые входные параметры} 
 Writeln('Введите количество промежутков в первой части пластины, 
N1'); 
 Readln(N1); 
 Writeln('Введите количество промежутков во второй части пластины, 
N2'); 
 Readln(N2); 
 Writeln('Введите окончание по времени, t_end'); 
 Readln(t_end); 
 Writeln('Введите толщину пластины, L'); 
 Readln(L); 
 Writeln('Введите коэффициент теплопроводности материала первой 
части пластины, lamda1'); 
 Readln(lamda1); 
 Writeln('Введите коэффициент теплопроводности материала второй 
части пластины, lamda2'); 
 Readln(lamda2); 
 Writeln('Введите плотность материала первой части пластины, ro1'); 
 Readln(ro1); 
 Writeln('Введите плотность материала второй части пластины, ro2'); 
 Readln(ro2); 
 Writeln('Введите теплоемкость материала первой части пластины, c1'); 



 54

 Readln(c1); 
 Writeln('Введите теплоемкость материала второй части пластины, c2'); 
 Readln(c2); 
 Writeln('Введите температуру на границе х = 0, Tl'); 
 Readln(Tl); 
 Writeln('Введите температуру на границе х = L, Tr'); 
 Readln(Tr); 
 Writeln('Введите начальную температуру, T0'); 
 Readln(T0); 
{определяем общее число узлов в пластине} 
 N:=N1+N2+1; 
{определяем расчетный шаг сетки по пространственной координате} 
 h:=L/(N-1); 
{определяем коэффициенты температуропроводности} 
 a1:=lamda1/(ro1*c1); 
 a2:=lamda2/(ro2*c2); 
{определяем расчетный шаг сетки по времени} 
 tau:=t_end/100.0; 
{определяем поле температуры в начальный момент времени} 
 for i:= 1 to N do 
  T[i]:=T0; 
{проводим интегрирование нестационарного уравнения 
теплопроводности} 
 time:=0; 
 while time<t_end do {используем цикл с предусловием} 
  begin 
{увеличиваем переменную времени на шаг τ} 
   time:=time+tau; 
{определяем начальные прогоночные коэффициенты на основе левого 
граничного условия} 
   alfa[1]:=0.0; 
   beta[1]:=Tl; 
{цикл с параметром для определения прогоночных коэффициентов по 
формуле (8) в первой части пластины} 
   for i:= 2 to N1 do 
    begin 
{ai, bi, ci, fi – коэффициенты канонического представления СЛАУ с 
трехдиагональной матрицей} 
     ai:=lamda1/sqr(h); 
     bi:=2.0*lamda1/sqr(h)+ro1*c1/tau; 
     ci:=lamda1/sqr(h); 

     fi:=-ro1*c1*T[i]/tau; 
{alfa[i], beta[i] – прогоночные коэффициенты} 
     alfa[i]:=ai/(bi-ci*alfa[i-1]); 
     beta[i]:=(ci*beta[i-1]-fi)/(bi-ci*alfa[i-1]); 
    end; 
{определяем прогоночные коэффициенты на границе раздела двух 
частей, используем соотношения (28)} 
   alfa[N1+1]:=2.0*a1*a2*tau*lamda2/(2.0*a1*a2*tau*(lamda2+lamda1 
                        *(1-alfa[N1]))+sqr(h)*(a1*lamda2+a2*lamda1)); 
   beta[N1+1]:=(2.0*a1*a2*tau*lamda1*beta[N1]+sqr(h)*(a1*lamda2+a2 
                        *lamda1)*T[N1+1])/(2.0*a1*a2*tau*(lamda2+lamda1 
                        *(1-alfa[N1]))+sqr(h)*(a1*lamda2+a2*lamda1)); 
{цикл с параметром для определения прогоночных коэффициентов по 
формуле (8) во второй части пластины} 
   for i:= N1+2 to N-1 do 
    begin 
{ai, bi, ci, fi – коэффициенты канонического представления СЛАУ с 
трехдиагональной матрицей} 
     ai:=lamda2/sqr(h); 
     bi:=2.0*lamda2/sqr(h)+ro2*c2/tau; 
     ci:=lamda2/sqr(h); 
     fi:=-ro2*c2*T[i]/tau; 
{alfa[i], beta[i] – прогоночные коэффициенты} 
     alfa[i]:=ai/(bi-ci*alfa[i-1]); 
     beta[i]:=(ci*beta[i-1]-fi)/(bi-ci*alfa[i-1]); 
    end; 
{определяем значение температуры на правой границе} 
   T[N]:=Tr; 
{используя соотношение (7) определяем неизвестное поле 
температуры} 
   for i:= N-1 downto 1 do 
    T[i]:=alfa[i]*T[i+1]+beta[i]; 
  end; {цикл с предусловием окончен} 
{выводим результат в файл} 
 Assign(f,'res.txt'); 
 Rewrite(f); 
 Writeln(f,'Толщина пластины L = ',L:6:4); 
 Writeln(f,'Толщина первой части пластины = ',N1*h:6:4); 
 Writeln(f,'Толщина второй части пластины = ',N2*h:6:4); 
 Writeln(f,'Число промежутков по координате в первой части пластины 
N1 = ',N1); 

 Writeln(f,'Число промежутков по координате во второй части пластины 
N2 = ',N2); 
 Writeln(f,'Общее число узлов по координате N = ',N); 
 Writeln(f,'Коэффициент теплопроводности материала первой части 
пластины lamda1 = ', lamda1:6:4); 
 Writeln(f,'Коэффициент теплопроводности материала второй части 
пластины lamda2 = ', lamda2:6:4); 
 Writeln(f,'Плотность материала первой части пластины ro1 = ',ro1:6:4); 
 Writeln(f,'Плотность материала второй части пластины ro2 = ',ro2:6:4); 
 Writeln(f,'Теплоемкость материала первой части пластины с1 = ',c1:6:4); 
 Writeln(f,'Теплоемкость материала второй части пластины с2 = ',c2:6:4); 
 Writeln(f,'Начальная температура  T0 = ',T0:6:4); 
 Writeln(f,'Температура на границе х = 0, Tl = ',Tl:6:4); 
 Writeln(f,'Температура на границе х = L, Tr = ',Tr:6:4); 
 Writeln(f,'Результат получен с шагом по координате h = ',h:6:4); 
 Writeln(f,'Результат получен с шагом по времени tau = ',tau:6:4); 
 Writeln(f,'Температурное поле в момент времени t = ',t_end:6:4); 
 close(f); 
 Assign(g,'tempr.txt'); 
 Rewrite(g); 
 for i:=1 to N do 
  writeln(g,'  ',h*(i-1):10:8,'   ',T[i]:8:5); 
 close(g); 
end. 

 Получены следующие распределения температуры: 
 

 
Рис. 15. Распределения температуры по толщине пластины в различные 

моменты времени 
 

2.5. ЗАДАЧА ТЕПЛОПРОВОДНОСТИ С ВНУТРЕННИМИ 
ИСТОЧНИКАМИ 

 
 Пусть в неограниченной пластине толщины м 3.0=L  действуют 
равномерно распределенные внутренние источники тепла мощностью 

серебро (λ = 419 Вт/(м⋅ºC), ρ = 10500 кг/м3, с = 200 Дж/(кг⋅ºC)). На 
границах 0=x  и Lx =  осуществляется теплообмен с окружающей 



 58

средой ( ( )СмВт 50 02 ⋅=κ , C60 0e =T ). Определим температурные 
поля через 1 с, 5 с и 10 с. 
 Математическая постановка задачи будет иметь вид: 

 Начальные и граничные условия запишутся следующим образом: 

 Для решения сформулированной краевой задачи применим метод 
конечных разностей на основе неявной четырехточечной схемы. В 
результате аппроксимации частных производных получаем следующую 

 Полученную систему можно свести к наиболее общему виду: 

 Прогоночные коэффициенты находятся по формулам (8). Далее 
неизвестное поле температуры определяется по выражению (7). 

Блок-схема к рассматриваемой задаче имеет вид: 

Ниже приведен листинг программы для решения рассматриваемой 
задачи (на языке программирования Pascal) 

 
uses crt; 
const mf=500; 
type 
     vector=array[1..mf] of real; 
var {раздел описания переменных, которые мы будем использовать в 
программе} 
 i, j, N, N1, N2, N3         : integer; 
 T, alfa, beta                   : vector; 
 ai, bi, ci, fi                     : real; 
 a, lamda, ro, c               : real; 
 h, tau, t_end, time         : real; 
 T0, L, kapa, Te, q         : real; 
 f, g                                 : text; 
begin 
 clrscr; 
{с клавиатуры вводим все необходимые входные параметры} 
 Writeln('Введите количество пространственных узлов в пластине, N'); 
 Readln(N); 
 Writeln('Введите окончание по времени, t_end'); 
 Readln(t_end); 
 Writeln('Введите толщину пластины, L'); 
 Readln(L); 
 Writeln('Введите коэффициент теплопроводности материала пластины, 
lamda'); 
 Readln(lamda); 
 Writeln('Введите плотность материала пластины, ro'); 
 Readln(ro); 
 Writeln('Введите теплоемкость материала пластины, c'); 
 Readln(c); 
 Writeln('Введите температуру окружающей среды, Te'); 
 Readln(Te); 
 Writeln('Введите коэффициент теплообмена, kapa'); 
 Readln(kapa); 
 Writeln('Введите начальную температуру, T0'); 
 Readln(T0); 
 Writeln('Введите мощность внутренних источников тепла, q'); 
 Readln(q); 

{определяем номера узлов, в которых расположен источник} 
 N1:=N div 4; 
 N2:=2*N1; 
 N3:=3*N1; 
{определяем расчетный шаг сетки по пространственной координате} 
 h:=L/(N-1); 
{определяем коэффициент температуропроводности} 
 a:=lamda/(ro*c); 
{определяем расчетный шаг сетки по времени} 
 tau:=t_end/100.0; 
{определяем поле температуры в начальный момент времени} 
 for i:= 1 to N do 
  T[i]:=T0; 
{проводим интегрирование нестационарного уравнения 
теплопроводности} 
 time:=0; 
 while time<t_end do {используем цикл с предусловием} 
  begin 
{увеличиваем переменную времени на шаг τ} 
   time:=time+tau; 
{определяем начальные прогоночные коэффициенты на основе левого 
граничного условия, используя соотношения (24)} 
   alfa[1]:=2.0*a*tau*lamda/(lamda*sqr(h)+2.0*a*tau*(lamda+kapa*h)); 
  beta[1]:=(lamda*sqr(h)*T[1]+2.0*a*tau*kapa*h*Te)/(lamda*sqr(h)+2.0 
                 *a*tau*(lamda+kapa*h)); 
{цикл с параметром для определения прогоночных коэффициентов по 
формуле (8)} 
   for i:= 2 to N-1 do 
    begin 
{ai, bi, ci, fi – коэффициенты канонического представления системы 
уравнений с трехдиагональной матрицей} 
     ai:=lamda/sqr(h); 
     bi:=2.0*lamda/sqr(h)+ro*c/tau; 
     ci:=lamda/sqr(h); 
     fi:=-ro*c*T[i]/tau; 
{определяем fi в зависимости от рассматрвиаемой точки 
пространства} 
     if (i=N1) or (i=N2) then fi:=-ro*c*T[i]/tau-h*(i-1)*q; 
     if i=N3 then fi:=-ro*c*T[i]/tau-h*(N1-1)*q; 
{alfa[i], beta[i] – прогоночные коэффициенты} 
     alfa[i]:=ai/(bi-ci*alfa[i-1]); 

     beta[i]:=(ci*beta[i-1]-fi)/(bi-ci*alfa[i-1]); 
    end; 
{определяем значение температуры на правой границе, используя 
соотношение (25)} 
   T[N]:=(lamda*sqr(h)*T[N]+2.0*a*tau*(lamda*beta[N-1]+kapa*h*Te)) 
              /(lamda*sqr(h)+2.0*a*tau*(lamda*(1-alfa[N-1])+kapa*h)); 
{используя соотношение (7) определяем неизвестное поле 
температуры} 
   for i:= N-1 downto 1 do 
    T[i]:=alfa[i]*T[i+1]+beta[i]; 
  end; {цикл с предусловием окончен} 
{выводим результат в файл} 
 Assign(f,'res.txt'); 
 Rewrite(f); 
 Writeln(f,'Толщина пластины L = ',L:6:4); 
 Writeln(f,'Число узлов по пространственной координате в пластине N = 
',N); 
 Writeln(f,'Внутренние источники находятся в точках = ',N1,'  ',N2,'  ',N3); 
 Writeln(f,'Внутренние источники находятся в точках = ',(N1-1)*h:6:4,'  
',(N2-1)*h:6:4,'  ', (N3-1)*h:6:4); 
 Writeln(f,'Коэффициент теплопроводности материала пластины lamda = 
',lamda:6:4); 
 Writeln(f,'Плотность материала пластины ro = ',ro:6:4); 
 Writeln(f,'Теплоемкость материала пластины с = ',c:6:4); 
 Writeln(f,'Начальная температура T0 = ',T0:6:4); 
 Writeln(f,'Коэффициент теплообмена kapa = ',kapa:6:4); 
 Writeln(f,'Температура окружающей среды Te = ',Te:6:4); 
 Writeln(f,'Составляющая мощности внутренних источников тепла q = 
',q:6:4); 
 Writeln(f,'Результат получен с шагом по координате h = ',h:6:4); 
 Writeln(f,'Результат получен с шагом по времени tau = ',tau:6:4); 
 Writeln(f,'Температурное поле в момент времени t = ',t_end:6:4); 
 close(f); 
 Assign(g,'tempr.txt'); 
 Rewrite(g); 
 for i:=1 to N do 
  writeln(g,'  ',h*(i-1):10:8,'   ',T[i]:8:5); 
 close(g); 
end. 

 Получены следующие распределения температуры: 
 

 
Рис. 16. Распределения температуры по толщине пластины в различные 

моменты времени 
 

2.6. ДВУМЕРНАЯ ЗАДАЧА ТЕПЛОПРОВОДНОСТИ ДЛЯ 
ОДНОРОДНОГО ТЕЛА 

 
 Проанализируем процесс теплопереноса в пластине (рис. 17). 
 

 
Рис. 17. Область решения 

 Медная пластина с размерами м5.0== HL . Горизонтальные 
границы являются адиабатическими, а на вертикальных границах 
поддерживаются постоянные температуры C 80 0=hT  и C 30 0=сT . 
Начальная температура области решения C 5 0

0 =T . 
 Математическая постановка задачи будет иметь вид: 

.0 
;0 

   ,2

2

2

2

Hy
Lx

y
T

x
T

t
Tc

<<
<<

⎟⎟
⎠

⎞
⎜⎜
⎝

⎛

∂
∂

+
∂
∂

λ=
∂
∂

ρ     (30) 

 Начальные и граничные условия запишутся следующим образом: 

 Для аппроксимации дифференциального уравнения (30) 
разностным введем пространственно-временную сетку с координатами 

( ) ( ) ,     ,1     ,1 τ⋅=⋅−=⋅−= nthjyhix nyjxi  
где yx hh   ,  – шаги сетки по координатам х, у соответственно; τ – шаг по 

времени; .,0   ;,1   ;,1 KnNjNi yx ===  Т.е. вся расчетная область       
(рис. 17) покрывается сеткой (рис. 18). 
 

 
Рис. 18. Разностная сетка области решения 

 Введем следующее обозначение: ( ) .,, ,
n
jinji TtyxT =  

 Дискретизацию уравнения (30) будем проводить на основе 
локально одномерной схемы А.А. Самарского [2], которая является 
абсолютно устойчивой и обладает свойством суммарной 
аппроксимации. Сущность этого подхода состоит в том, что шаг по 
времени реализуется в два этапа – на промежуточном временном шаге 
проводим дискретизацию двумерного уравнения (30) только в 
направлении оси х и получаем одномерное уравнение, после его 
решения проводим вновь дискретизацию уравнения (30), но уже в 
направлении оси у и, решая полученное одномерное уравнение, 
определяем поле температуры на целом шаге по времени. 
 Итак: 

 Разностные уравнения (32), (33) сводятся к стандартному 
трехдиагональному виду и решаются последовательно методом 
прогонки (пункт 2.1.). Сначала для всей области решается уравнение 
(32), после того как его решение будет найдено, переходят к решению 
уравнения (33). 
 Рассмотрим решение уравнения (32) методом прогонки. Приведем 

 Для определения прогоночных коэффициентов по соотношению 
(8) необходимо найти 11  и βα  из левого граничного условия. Далее 

определяя значение 2
1

,

+n

jNx
T  из правого граничного условия, находят поле 

температуры 2
1

,

+n

jiT  на промежуточном временном слое по формулам (7). 
После этого приступают к решению уравнения (33). Этапы решения 
уравнения (33) аналогичны решению уравнения (32). 

Блок-схема к рассматриваемой задаче имеет вид: 

Ниже приведен листинг программы для решения рассматриваемой 
задачи (на языке программирования Pascal) 

 
uses crt; 
const mf=102; 
type 
     vector1=array[1..mf] of real; 
     vector2=array[1..mf,1..mf] of real; 
var {раздел описания переменных, которые мы будем использовать в 
программе} 
 i, j, Nx, Ny                      : integer; 
 T                                      : vector2; 
 alfa, beta                         : vector1; 
 ai, bi, ci, fi                       : real; 
 a, lamda, ro, c                 : real; 
 hx, hy, tau, t_end, time   : real; 
 T0, L, H, Th, Tc              : real; 
 f, g                                   : text; 
begin 
 clrscr; 
{с клавиатуры вводим все необходимые входные параметры} 
 Writeln('Введите количество пространственных узлов в пластине по оси 
х, Nx'); 
 Readln(Nx); 
 Writeln('Введите количество пространственных узлов в пластине по оси 
y, Ny'); 
 Readln(Ny); 
 Writeln('Введите окончание по времени, t_end'); 
 Readln(t_end); 
 Writeln('Введите длину пластины, L'); 
 Readln(L); 
 Writeln('Введите толщину пластины, H'); 
 Readln(H); 
 Writeln('Введите коэффициент теплопроводности материала пластины, 
lamda'); 
 Readln(lamda); 
 Writeln('Введите плотность материала пластины, ro'); 
 Readln(ro); 
 Writeln('Введите теплоемкость материала пластины, c'); 
 Readln(c); 
 Writeln('Введите температуру на границе х = 0 области решения, Th'); 

 Readln(Th); 
 Writeln('Введите температуру на границе х = L области решения, Tc'); 
 Readln(Tc); 
 Writeln('Введите начальную температуру, T0'); 
 Readln(T0); 
{определяем расчетные шаги сетки по пространственным 
координатам} 
 hx:=L/(Nx-1); 
 hy:=H/(Ny-1); 
{определяем коэффициент температуропроводности} 
 a:=lamda/(ro*c); 
{определяем расчетный шаг сетки по времени} 
 tau:=t_end/100.0; 
{определяем поле температуры в начальный момент времени} 
 for i:= 1 to Nx do 
  for j:= 1 to Ny do 
    T[i,j]:=T0; 
{проводим интегрирование нестационарного уравнения 
теплопроводности} 
 time:=0; 
 while time<t_end do {используем цикл с предусловием} 
  begin 
{увеличиваем переменную времени на шаг τ} 
   time:=time+tau; 
{решаем СЛАУ в направлении оси Ох для определения поля 
температуры на промежуточном временном слое} 
   for j:=1 to Ny do 
    begin 
{определяем начальные прогоночные коэффициенты на основе левого 
граничного условия} 
     alfa[1]:=0.0; 
     beta[1]:=Th; 
{цикл с параметром для определения прогоночных коэффициентов по 
формуле (8)} 
     for i:= 2 to Nx-1 do 
      begin 
{ai, bi, ci, fi – коэффициенты канонического представления СЛАУ с 
трехдиагональной матрицей} 
       ai:=lamda/sqr(hx); 
       bi:=2.0*lamda/sqr(hx)+ro*c/tau; 
       ci:=lamda/sqr(hx); 

       fi:=-ro*c*T[i,j]/tau; 
{alfa[i], beta[i] – прогоночные коэффициенты} 
       alfa[i]:=ai/(bi-ci*alfa[i-1]); 
       beta[i]:=(ci*beta[i-1]-fi)/(bi-ci*alfa[i-1]); 
      end; 
{определяем значение температуры на правой границе на основе 
правого граничного условия} 
     T[Nx,j]:=Tc; 
{используя соотношение (7) определяем неизвестное поле температуры 
на промежуточном временном слое} 
     for i:= Nx-1 downto 1 do 
      T[i,j]:=alfa[i]*T[i+1,j]+beta[i]; 
    end; 
{решаем СЛАУ в направлении оси Оу для определения поля 
температуры на целом временном слое} 
   for i:=2 to Nx-1 do 
    begin 
{определяем начальные прогоночные коэффициенты на основе нижнего 
граничного условия, используя соотношения (20) при условии, что         
q1 = 0} 
     alfa[1]:=2.0*a*tau/(2.0*a*tau+sqr(hy)); 
     beta[1]:=sqr(hy)*T[i,1]/(2.0*a*tau+sqr(hy)); 
{цикл с параметром для определения прогоночных коэффициентов по 
формуле (8)} 
     for j:= 2 to Ny-1 do 
      begin 
{ai, bi, ci, fi – коэффициенты канонического представления СЛАУ с 
трехдиагональной матрицей} 
       ai:=lamda/sqr(hy); 
       bi:=2.0*lamda/sqr(hy)+ro*c/tau; 
       ci:=lamda/sqr(hy); 
       fi:=-ro*c*T[i,j]/tau; 
{alfa[j], beta[j] – прогоночные коэффициенты} 
       alfa[j]:=ai/(bi-ci*alfa[j-1]); 
       beta[j]:=(ci*beta[j-1]-fi)/(bi-ci*alfa[j-1]); 
      end; 
{определяем значение температуры на верхней границе, используя 
соотношение (21) при условии, что q2 = 0} 
     T[i,Ny]:=(2.0*a*tau*beta[Ny-1]+sqr(hy)*T[i,Ny])/(2.0*a*tau 
                     *(1.0-alfa[Ny-1])+sqr(hy)); 

{используя соотношение (7) определяем неизвестное поле температуры 
на промежуточном временном слое} 
     for j:= Ny-1 downto 1 do 
      T[i,j]:=alfa[j]*T[i,j+1]+beta[j]; 
    end; 
  end; {цикл с предусловием окончен} 
{выводим результат в файл} 
 Assign(f,'res.txt'); 
 Rewrite(f); 
 Writeln(f,'Длина пластины L = ',L:6:4); 
 Writeln(f,'Толщина пластины H = ',H:6:4); 
 Writeln(f,'Число узлов по пространственной координате x в пластине 
Nx = ',Nx); 
 Writeln(f,'Число узлов по пространственной координате y в пластине 
Ny = ',Ny); 
 Writeln(f,'Коэффициент теплопроводности материала пластины lamda = 
',lamda:6:4); 
 Writeln(f,'Плотность материала пластины ro = ',ro:6:4); 
 Writeln(f,'Теплоемкость материала пластины с = ',c:6:4); 
 Writeln(f,'Начальная температура T0 = ',T0:6:4); 
 Writeln(f,'Температура на границе x = 0 области решения Th = ',Th:6:4); 
 Writeln(f,'Температура на границе x = L области решения Tc = ',Tc:6:4); 
 Writeln(f,'Результат получен с шагом по координате x hx = ',hx:6:4); 
 Writeln(f,'Результат получен с шагом по координате y hy = ',hy:6:4); 
 Writeln(f,'Результат получен с шагом по времени tau = ',tau:6:4); 
 Writeln(f,'Температурное поле в момент времени t = ',t_end:6:4); 
 close(f); 
 Assign(g,'tempr.txt'); 
 Rewrite(g); 
 for i:=1 to Nx do 
  for j:=1 to Ny do 
   writeln(g,'  ',hx*(i-1):10:8,'  ',hy*(j-1):10:8,'   ',T[i,j]:8:5); 
 close(g); 
end. 

 Получено следующее распределение температуры: 
 

 
Рис. 19. Изотермы (ºС) в пластине при t = 60 c 

 
2.7. ДВУМЕРНАЯ ЗАДАЧА ТЕПЛОПРОВОДНОСТИ ДЛЯ 

НЕОДНОРОДНОГО ТЕЛА 
 
 Проанализируем процесс теплопереноса в пластине, содержащей 
два включения (рис. 20). Рассматриваемая задача является учебной, 
поэтому определяющие размеры 1l , 2l , 3l , 4l , 1h , 2h , 3h , 4h  выбираются 
таким образом, чтобы используемая разностная сетка была 
равномерной. Для этой цели в программе в качестве входных 
параметров будем задавать не линейные размеры, а количество 
промежутков, характеризующих рассматриваемый отрезок. 

Рис. 20. Область решения 

 
 Медная пластина (1 на рис. 20) с размерами м 5.0== HL . 
Материалы включений: сталь (2 на рис. 20) (λ2 = 46 Вт/(м⋅ºC), ρ2 = 7800 
кг/м3, с2 = 460 Дж/(кг⋅ºC)) и железо (3 на   рис. 20) (λ3 = 71 Вт/(м⋅ºC), ρ3 = 
7900 кг/м3, с3 = 460 Дж/(кг⋅ºC)). На вертикальных границах области 
решения поддерживаются постоянные температуры C 100 0=hT  при 0=x  
и C 0 0=cT  при Lx = . Горизонтальные границы являются 
адиабатическими. Начальная температура области решения C 50 0

0 =T . 

 Математическая постановка задачи будет иметь вид: 

 Начальные и граничные условия запишутся следующим образом: 

 Рассматриваемая задача представляет собой своеобразную 
комбинацию задач 2.4. и 2.6. Для решения сформулированной задачи 
(34)–(36) также как и в 2.6. введем равномерную пространственно-
временную сетку. 
 Дискретизацию уравнений (34) будем проводить на основе 
локально одномерной схемы А.А. Самарского. Решение полученных 
систем линейных алгебраических уравнений проводится методом 
прогонки, при этом необходимо учесть, что в пластине 1 присутствуют 
две неоднородности. Эти включения учитываются в прогоночных 
коэффициентах на границах сопряжения, а также в коэффициентах 
канонического уравнения вида (6) в зависимости от материала элемента. 
 Алгоритм решения рассматриваемой задачи можно представить 
следующим образом. 
 Сначала поэтапно решается СЛАУ вида (32), т.е. всю область 
решения делим на однородные части, например, 

,  ,  ,0 321212111 hhhyhhhhyhhy ++<<++<<<≤  
.  , 43214321321 Hyhhhhhhhhyhhh ≤<++++++<<++  И для каждого 

такого участка решаем систему уравнений вида (32), как для случая 
одномерной задачи теплопроводности двухслойной пластины        
(пункт 2.4.). На границах же вида 43213211   , llllxlllhy +++<<++=  в 
качестве теплофизических параметров среды используем среднее 
арифметическое, в данном случае, 

.
2

   ,
2

   ,
2

313131 сссэффэффэфф
+

=
ρ+ρ

=ρ
λ+λ

=λ  При этом на данной 

границе в точке 1321   , hylllx =++= , справа рассматривается среда с 
параметрами эффэффэфф с , ,ρλ , а слева – 111  , , сρλ . После такого решения 
системы вида (32) переходят к решению системы вида (33), которая 
разрешается аналогично. 
 

Ниже приведен листинг программы для решения рассматриваемой 
задачи (на языке программирования Pascal) 

 
uses crt; 
const mf=101; 
type 
     vector1=array[1..mf] of real; 
     vector2=array[1..mf,1..mf] of real; 

var {раздел описания переменных, которые мы будем использовать в 
программе} 
 i, j, Nx, Ny                     : integer; 
 nx1, nx2, nx3, nx4         : integer; 
 ny1, ny2, ny3, ny4         : integer; 
 T                                     : vector2; 
 a1, lamda1, ro1, c1        : real; 
 a2, lamda2, ro2, c2        : real; 
 a3, lamda3, ro3, c3        : real; 
 hx, hy, tau, t_end, time  : real; 
 T0, L, H, Th, Tc            : real; 
 f, g                                 : text; 
 
procedure progonx(j: integer; lamda,ro,c:real; var W:vector2); 
{процедура, разрешающая СЛАУ с трехдиагональной матрицей 
методом прогонки, в направлении оси Ох 
j – номер слоя по оси у вдоль которого происходит решение СЛАУ; 
lamda – коэффициент теплопроводности; 
ro – плотность; 
c – коэффициент теплоемкости; 
W – двумерное поле температуры} 
var {раздел описания локальных переменных} 
  i                    : integer; 
  alfa, beta       : vector1; 
  ai, bi, ci, fi    : real; 
begin 
{определяются начальные прогоночные коэффициенты на основе левого 
граничного условия} 
 alfa[1]:=0.0; 
 beta[1]:=Th; 
{цикл с параметром для определения прогоночных коэффициентов по 
формуле (8)} 
 for i:= 2 to Nx-1 do 
  begin 
{ai, bi, ci, fi – коэффициенты канонического представления СЛАУ с 
трехдиагональной матрицей} 
   ai:=lamda/sqr(hx); 
   bi:=2.0*lamda/sqr(hx)+ro*c/tau; 
   ci:=lamda/sqr(hx); 
   fi:=-ro*c*W[i,j]/tau; 

{alfa[i], beta[i] – прогоночные коэффициенты} 
   alfa[i]:=ai/(bi-ci*alfa[i-1]); 
   beta[i]:=(ci*beta[i-1]-fi)/(bi-ci*alfa[i-1]); 
  end; 
{определяем значение температуры на правой границе на основе 
правого граничного условия} 
 W[Nx,j]:=Tc; 
{используя соотношение (7) определяем неизвестное поле 
температуры} 
 for i:= Nx-1 downto 1 do 
  W[i,j]:=alfa[i]*W[i+1,j]+beta[i]; 
end; {окончание процедуры progonx} 
 
procedure progony(i:integer; lamda,ro,c:real; var W:vector2); 
{процедура, разрешающая СЛАУ с трехдиагональной матрицей 
методом прогонки, в направлении оси Оу 
i – номер слоя по оси x вдоль которого происходит решение СЛАУ; 
lamda – коэффициент теплопроводности; 
ro – плотность; 
c – коэффициент теплоемкости; 
W – двумерное поле температуры} 
var {раздел описания локальных переменных} 
  j                       : integer; 
  alfa, beta          : vector1; 
  a, ai, bi, ci, fi   : real; 
begin 
{определяется коэффициент температуропроводности} 
 a:=lamda/(ro*c); 
{определяются начальные прогоночные коэффициенты на основе 
нижнего граничного условия, в соответствии с рассматриваемой 
постановкой используется соотношение (20)} 
 alfa[1]:=2.0*a*tau/(2.0*a*tau+sqr(hy)); 
 beta[1]:=sqr(hy)*W[i,1]/(2.0*a*tau+sqr(hy)); 
{цикл с параметром для определения прогоночных коэффициентов по 
формуле (8)} 
 for j:= 2 to Ny-1 do 
  begin 
{ai, bi, ci, fi – коэффициенты канонического представления СЛАУ с 
трехдиагональной матрицей} 
   ai:=lamda/sqr(hy); 
   bi:=2.0*lamda/sqr(hy)+ro*c/tau; 

   ci:=lamda/sqr(hy); 
   fi:=-ro*c*W[i,j]/tau; 
{alfa[j], beta[j] – прогоночные коэффициенты} 
   alfa[j]:=ai/(bi-ci*alfa[j-1]); 
   beta[j]:=(ci*beta[j-1]-fi)/(bi-ci*alfa[j-1]); 
  end; 
{определяем значение температуры на верхней границе на основе 
верхнего граничного условия, в нашем случае используется 
соотношение (21)} 
 W[i,Ny]:=(2.0*a*tau*lamda*beta[Ny-1]+lamda*sqr(hy)*W[i,Ny]) 
                  /(lamda*sqr(hy)+2.0*a*tau*lamda*(1.0-alfa[Ny-1])); 
{используя соотношение (7) определяем неизвестное поле 
температуры} 
 for j:= Ny-1 downto 1 do 
  W[i,j]:=alfa[j]*W[i,j+1]+beta[j]; 
end; {окончание процедуры progonу} 
 
procedure progonxIV(j,n1,n2:integer; lamda,lamdan,ro,ron,c,cn:real; var 
W:vector2); 
{процедура, разрешающая СЛАУ с трехдиагональной матрицей 
методом прогонки, в направлении оси Оx при наличии включения с 
отличающимися теплофизическими характеристиками 
j – номер слоя по оси у вдоль которого происходит решение СЛАУ; 
n1 – номер узла с которого начинается включение; 
n2 – номер узла которым заканчивается включение; 
lamda – коэффициент теплопроводности основного материала; 
lamdan – коэффициент теплопроводности материала включения; 
ro – плотность основного материала; 
ron – плотность материала включения; 
c – коэффициент теплоемкости основного материала; 
cn – коэффициент теплоемкости материала включения; 
W – двумерное поле температуры} 
var {раздел описания локальных переменных} 
  i                   : integer; 
  alfa, beta      : vector1; 
  ai, bi, ci, fi   : real; 
  a, an             : real; 
begin 
{определяются коэффициенты температуропроводности} 
 a:=lamda/(ro*c); 
 an:=lamdan/(ron*cn); 

{определяются начальные прогоночные коэффициенты на основе левого 
граничного условия} 
 alfa[1]:=0.0; 
 beta[1]:=Th; 
{цикл с параметром для определения прогоночных коэффициентов по 
формуле (8) до включения} 
 for i:= 2 to n1 do 
  begin 
{ai, bi, ci, fi – коэффициенты канонического представления СЛАУ с 
трехдиагональной матрицей} 
   ai:=lamda/sqr(hx); 
   bi:=2.0*lamda/sqr(hx)+ro*c/tau; 
   ci:=lamda/sqr(hx); 
   fi:=-ro*c*W[i,j]/tau; 
{alfa[i], beta[i] – прогоночные коэффициенты} 
   alfa[i]:=ai/(bi-ci*alfa[i-1]); 
   beta[i]:=(ci*beta[i-1]-fi)/(bi-ci*alfa[i-1]); 
  end; 
{определяются прогоночные коэффициенты на границе основного 
материала и включения} 
 alfa[n1+1]:=2.0*a*an*tau*lamdan/(2.0*a*an*tau*(lamdan+lamda 
                     *(1-alfa[n1]))+sqr(hx)*(a*lamdan+an*lamda)); 
beta[n1+1]:=(2.0*a*an*tau*lamda*beta[n1]+sqr(hx)*(a*lamdan+an*lamda) 
                     *W[n1+1,j])/(2.0*a*an*tau*(lamdan+lamda*(1-alfa[n1])) 
                     +sqr(hx)*(a*lamdan+an*lamda)); 
{цикл с параметром для определения прогоночных коэффициентов по 
формуле (8) во включении} 
 for i:= n1+2 to n1+n2 do 
  begin 
{ai, bi, ci, fi – коэффициенты канонического представления СЛАУ с 
трехдиагональной матрицей} 
   ai:=lamdan/sqr(hx); 
   bi:=2.0*lamdan/sqr(hx)+ron*cn/tau; 
   ci:=lamdan/sqr(hx); 
   fi:=-ron*cn*W[i,j]/tau; 
{alfa[i], beta[i] – прогоночные коэффициенты} 
   alfa[i]:=ai/(bi-ci*alfa[i-1]); 
   beta[i]:=(ci*beta[i-1]-fi)/(bi-ci*alfa[i-1]); 
  end; 

{определяются прогоночные коэффициенты на границе включения и 
основного материала} 
  alfa[n1+n2+1]:=2.0*an*a*tau*lamda/(2.0*an*a*tau*(lamda+lamdan* 
                            (1-alfa[n1+n2]))+sqr(hx)*(an*lamda+a*lamdan)); 
  beta[n1+n2+1]:=(2.0*an*a*tau*lamdan*beta[n1+n2]+sqr(hx)*(a*lamdan 
                            +an*lamda)*W[n1+n2+1,j])/(2.0*an*a*tau*(lamda 
                            +lamdan*(1-alfa[n1+n2]))+sqr(hx)*(an*lamda+a*lamdan)); 
{цикл с параметром для определения прогоночных коэффициентов по 
формуле (8) в основном материале} 
 for i:= n1+n2+2 to Nx-1 do 
  begin 
{ai, bi, ci, fi – коэффициенты канонического представления СЛАУ с 
трехдиагональной матрицей} 
   ai:=lamda/sqr(hx); 
   bi:=2.0*lamda/sqr(hx)+ro*c/tau; 
   ci:=lamda/sqr(hx); 
   fi:=-ro*c*W[i,j]/tau; 
{alfa[i], beta[i] – прогоночные коэффициенты} 
   alfa[i]:=ai/(bi-ci*alfa[i-1]); 
   beta[i]:=(ci*beta[i-1]-fi)/(bi-ci*alfa[i-1]); 
  end; 
{определяем значение температуры на правой границе на основе 
правого граничного условия} 
 W[Nx,j]:=Tc; 
{используя соотношение (7) определяем неизвестное поле 
температуры} 
 for i:= Nx-1 downto 1 do 
  W[i,j]:=alfa[i]*W[i+1,j]+beta[i]; 
end; {окончание процедуры progonхIV} 
 
procedure progonyIV(i,n1,n2:integer; lamda,lamdan,ro,ron,c,cn:real; var 
W:vector2); 
{процедура, разрешающая СЛАУ с трехдиагональной матрицей 
методом прогонки, в направлении оси Оy при наличии включения с 
отличающимися теплофизическими характеристиками 
i – номер слоя по оси x вдоль которого происходит решение СЛАУ; 
n1 – номер узла с которого начинается включение; 
n2 – номер узла которым заканчивается включение; 
lamda – коэффициент теплопроводности основного материала; 
lamdan – коэффициент теплопроводности материала включения; 
ro – плотность основного материала; 

ron – плотность материала включения; 
c – коэффициент теплоемкости основного материала; 
cn – коэффициент теплоемкости материала включения; 
W – двумерное поле температуры} 
Var {раздел описания локальных переменных} 
  j                  : integer; 
  alfa, beta     : vector1; 
  ai, bi, ci, fi  : real; 
  a, an            : real; 
begin 
{определяются коэффициенты температуропроводности} 
 a:=lamda/(ro*c); 
 an:=lamdan/(ron*cn); 
{определяются начальные прогоночные коэффициенты на основе 
нижнего граничного условия, в данном случае используем соотношение 
(20)} 
 alfa[1]:=2.0*a*tau/(2.0*a*tau+sqr(hy)); 
 beta[1]:=sqr(hy)*W[i,1]/(2.0*a*tau+sqr(hy)); 
{цикл с параметром для определения прогоночных коэффициентов по 
формуле (8) до включения} 
 for j:= 2 to n1 do 
  begin 
{ai, bi, ci, fi – коэффициенты канонического представления СЛАУ с 
трехдиагональной матрицей} 
   ai:=lamda/sqr(hy); 
   bi:=2.0*lamda/sqr(hy)+ro*c/tau; 
   ci:=lamda/sqr(hy); 
   fi:=-ro*c*W[i,j]/tau; 
{alfa[j], beta[j] – прогоночные коэффициенты} 
   alfa[j]:=ai/(bi-ci*alfa[j-1]); 
   beta[j]:=(ci*beta[j-1]-fi)/(bi-ci*alfa[j-1]); 
  end; 
{определяются прогоночные коэффициенты на границе основного 
материала и включения} 
 alfa[n1+1]:=2.0*a*an*tau*lamdan/(2.0*a*an*tau*(lamdan+lamda 
                     *(1-alfa[n1]))+sqr(hy)*(a*lamdan+an*lamda)); 
 beta[n1+1]:=(2.0*a*an*tau*lamda*beta[n1]+sqr(hy)*(a*lamdan 
                      +an*lamda)*W[i,n1+1])/(2.0*a*an*tau*(lamdan+lamda 
                      *(1-alfa[n1]))+sqr(hy)*(a*lamdan+an*lamda)); 

{цикл с параметром для определения прогоночных коэффициентов по 
формуле (8) во включении} 
 for j:= n1+2 to n1+n2 do 
  begin 
{ai, bi, ci, fi – коэффициенты канонического представления СЛАУ с 
трехдиагональной матрицей} 
   ai:=lamdan/sqr(hy); 
   bi:=2.0*lamdan/sqr(hy)+ron*cn/tau; 
   ci:=lamdan/sqr(hy); 
   fi:=-ron*cn*W[i,j]/tau; 
{alfa[j], beta[j] – прогоночные коэффициенты} 
   alfa[j]:=ai/(bi-ci*alfa[j-1]); 
   beta[j]:=(ci*beta[j-1]-fi)/(bi-ci*alfa[j-1]); 
  end; 
{определяются прогоночные коэффициенты на границе включения и 
основного материала} 
 alfa[n1+n2+1]:=2.0*an*a*tau*lamda/(2.0*an*a*tau*(lamda+lamdan 
                           *(1-alfa[n1+n2]))+sqr(hy)*(an*lamda+a*lamdan)); 
 beta[n1+n2+1]:=(2.0*an*a1*tau*lamdan*beta[n1+n2]+sqr(hy)*(a*lamdan 
                           +an*lamda)*W[i,n1+n2+1])/(2.0*an*a*tau*(lamda+lamdan 
                           *(1-alfa[n1+n2]))+sqr(hy)*(an*lamda+a*lamdan)); 
{цикл с параметром для определения прогоночных коэффициентов по 
формуле (8) в основном материале} 
 for j:= n1+n2+2 to Ny-1 do 
  begin 
{ai, bi, ci, fi – коэффициенты канонического представления СЛАУ с 
трехдиагональной матрицей} 
   ai:=lamda/sqr(hy); 
   bi:=2.0*lamda/sqr(hy)+ro*c/tau; 
   ci:=lamda/sqr(hy); 
   fi:=-ro*c*W[i,j]/tau; 
{alfa[j], beta[j] – прогоночные коэффициенты} 
   alfa[j]:=ai/(bi-ci*alfa[j-1]); 
   beta[j]:=(ci*beta[j-1]-fi)/(bi-ci*alfa[j-1]); 
  end; 
{определяем значение температуры на верхней границе на основе 
верхнего граничного условия, в данном случае используем соотношение 
(21)} 
 W[i,Ny]:=(2.0*a*tau*lamda*beta[Ny-1]+lamda*sqr(hy)*W[i,Ny]) 
                 /(lamda*sqr(hy)+2.0*a*tau*lamda*(1.0-alfa[Ny-1])); 

{используя соотношение (7) определяем неизвестное поле 
температуры} 
 for j:= Ny-1 downto 1 do 
  W[i,j]:=alfa[j]*W[i,j+1]+beta[j]; 
end; {окончание процедуры progonуIV} 
begin 
 clrscr; 
{с клавиатуры вводим все необходимые входные параметры} 
 Writeln('Введите длину пластины, L'); 
 Readln(L); 
 Writeln('Введите толщину пластины, H'); 
 Readln(H); 
 Writeln('Введите количество пространственных узлов в пластине по оси 
х, Nx'); 
 Readln(Nx); 
 Writeln('Введите количество промежутков между границей х=0 и 
включением 2 по оси х, nx1'); 
 Readln(nx1); 
 Writeln('Введите количество промежутков во включении 2 по оси х, 
nx2'); 
 Readln(nx2); 
 Writeln('Введите количество промежутков между включением 2 и 
включением 3 по оси х, nx3'); 
 Readln(nx3); 
 Writeln('Введите количество промежутков во включении 3 по оси х, 
nx4'); 
 Readln(nx4); 
 Writeln('Введите количество пространственных узлов в пластине по оси 
y, Ny'); 
 Readln(Ny); 
 Writeln('Введите количество промежутков между границей у=0 и 
включением 3 по оси y, ny1'); 
 Readln(ny1); 
 Writeln('Введите количество промежутков во включении 3 по оси y, 
ny2'); 
 Readln(ny2); 
 Writeln('Введите количество промежутков между включением 3 и 
включением 2 по оси y, ny3'); 
 Readln(ny3); 
 Writeln('Введите количество промежутков во включении 2 по оси y, 
ny4'); 

 Readln(ny4); 
 Writeln('Введите окончание по времени, t_end'); 
 Readln(t_end); 
 Writeln('Введите коэффициент теплопроводности материала пластины, 
lamda1'); 
 Readln(lamda1); 
 Writeln('Введите плотность материала пластины, ro1'); 
 Readln(ro1); 
 Writeln('Введите теплоемкость материала пластины, c1'); 
 Readln(c1); 
 Writeln('Введите коэффициент теплопроводности материала включения 
2, lamda2'); 
 Readln(lamda2); 
 Writeln('Введите плотность материала включения 2, ro2'); 
 Readln(ro2); 
 Writeln('Введите теплоемкость материала включения 2, c2'); 
 Readln(c2); 
 Writeln('Введите коэффициент теплопроводности материала включения 
3, lamda3'); 
 Readln(lamda3); 
 Writeln('Введите плотность материала включения 3, ro3'); 
 Readln(ro3); 
 Writeln('Введите теплоемкость материала включения 3, c3'); 
 Readln(c3); 
 Writeln('Введите температуру на границе х=0, Th'); 
 Readln(Th); 
 Writeln('Введите температуру на границе х=L, Tc'); 
 Readln(Tc); 
 Writeln('Введите начальную температуру, T0'); 
 Readln(T0); 
{определяем расчетные шаги сетки по пространственным 
координатам} 
 hx:=L/(Nx-1); 
 hy:=H/(Ny-1); 
{определяем расчетный шаг сетки по времени} 
 tau:=t_end/100.0; 
{записываем все входные параметры} 
 Assign(f,'res.txt'); 
 Rewrite(f); 
 Writeln(f,'Длина пластины L = ',L:6:4); 

 Writeln(f,'Расстояние между границей х=0 и включением 2 = 
',nx1*hx:6:4); 
 Writeln(f,'Длина включения 2 = ',nx2*hx:6:4); 
 Writeln(f,'Расстояние между включением 2 и включением 3 = 
',nx3*hx:6:4); 
 Writeln(f,'Длина включения 3 = ',nx4*hx:6:4); 
 Writeln(f,'Толщина пластины H = ',H:6:4); 
 Writeln(f,'Расстояние между границей y=0 и включением 3 = 
',ny1*hy:6:4); 
 Writeln(f,'Толщина включения 3 = ',ny2*hy:6:4); 
 Writeln(f,'Расстояние между включением 3 и включением 2 = 
',ny3*hy:6:4); 
 Writeln(f,'Толщина включения 2 = ',ny4*hy:6:4); 
 Writeln(f,'Число узлов по пространственной координате x в пластине 
Nx = ',Nx); 
 Writeln(f,'Число промежутков между границей х=0 и включением 2 = 
',nx1); 
 Writeln(f,'Число промежутков во включении 2 = ',nx2); 
 Writeln(f,'Число промежутков между включением 2 и включением 3 = 
',nx3); 
 Writeln(f,'Число промежутков во включении 3 = ',nx4); 
 Writeln(f,'Число узлов по пространственной координате y в пластине 
Ny = ',Ny); 
 Writeln(f,'Число промежутков между границей y=0 и включением 3 = 
',ny1); 
 Writeln(f,'Число промежутков во включении 3 = ',ny2); 
 Writeln(f,'Число промежутков между включением 3 и включением 2 = 
',ny3); 
 Writeln(f,'Число промежутков во включении 2 = ',ny4); 
 Writeln(f,'Коэффициент теплопроводности материала пластины lamda1 
= ',lamda1:6:4); 
 Writeln(f,'Плотность материала пластины ro1 = ',ro1:6:4); 
 Writeln(f,'Теплоемкость материала пластины с1 = ',c1:6:4); 
 Writeln(f,'Коэффициент теплопроводности материала включения 2 
lamda2 = ',lamda2:6:4); 
 Writeln(f,'Плотность материала включения 2 ro2 = ',ro2:6:4); 
 Writeln(f,'Теплоемкость материала включения 2 с2 = ',c2:6:4); 
 Writeln(f,'Коэффициент теплопроводности материала включения 3 
lamda3 = ',lamda3:6:4); 
 Writeln(f,'Плотность материала включения 3 ro3 = ',ro3:6:4); 
 Writeln(f,'Теплоемкость материала включения 3 с3 = ',c3:6:4); 

 Writeln(f,'Начальная температура T0 = ',T0:6:4); 
 Writeln(f,'Температура на границе х=0 области решения Th = ',Th:6:4); 
 Writeln(f,'Температура на границе х=L области решения Tc = ',Tc:6:4); 
 Writeln(f,'Результат получен с шагом по координате x hx = ',hx:6:4); 
 Writeln(f,'Результат получен с шагом по координате y hy = ',hy:6:4); 
 Writeln(f,'Результат получен с шагом по времени tau = ',tau:6:4); 
 Writeln(f,'Температурное поле в момент времени t = ',t_end:6:4); 
 close(f); 
{определяем коэффициент температуропроводности} 
 a1:=lamda1/(ro1*c1); 
 a2:=lamda2/(ro2*c2); 
 a3:=lamda3/(ro3*c3); 
{определяем поле температуры в начальный момент времени} 
 for i:= 1 to Nx do 
  for j:= 1 to Ny do 
    T[i,j]:=T0; 
{проводим интегрирование нестационарного уравнения 
теплопроводности} 
 time:=0; 
 while time<t_end do {используем цикл с предусловием} 
  begin 
{увеличиваем переменную времени на шаг τ} 
   time:=time+tau; 
{             СЛАУ в направлении оси Ох 
   в промежутке между границей y=0 и включением 3} 
   for j:=1 to ny1 do 
    progonx(j,lamda1,ro1,c1,T); 
{на нижней границе включения 3} 
progonxIV(ny1+1,nx1+nx2+nx3,nx4,lamda1,0.5*(lamda1+lamda3),ro1,0.5*(
ro1+ro3),c1,0.5*(c1+c3),T); 
{во включении 3} 
   for j:=ny1+2 to ny1+ny2 do 
    progonxIV(j,nx1+nx2+nx3,nx4,lamda1,lamda3,ro1,ro3,c1,c3,T); 
{на верхней границе включения 3} 
progonxIV(ny1+ny2+1,nx1+nx2+nx3,nx4,lamda1,0.5*(lamda1+lamda3),ro1,
0.5*(ro1+ro3),c1,0.5*(c1+c3),T); 
{между включением 3 и включением 2} 
   for j:=ny1+ny2+2 to ny1+ny2+ny3 do 
    progonx(j,lamda1,ro1,c1,T); 

{на нижней границе включения 2} 
progonxIV(ny1+ny2+ny3+1,nx1,nx2,lamda1,0.5*(lamda1+lamda2),ro1,0.5*(
ro1+ro2),c1,0.5*(c1+c2),T); 
{во включении 2} 
   for j:=ny1+ny2+ny3+2 to ny1+ny2+ny3+ny4 do 
    progonxIV(j,nx1,nx2,lamda1,lamda2,ro1,ro2,c1,c2,T); 
{на верхней границе включения 2} 
progonxIV(ny1+ny2+ny3+ny4+1,nx1,nx2,lamda1,0.5*(lamda1+lamda2),ro1,
0.5*(ro1+ro2),c1,0.5*(c1+c2),T); 
{между включением 2 и границей y = H} 
   for j:=ny1+ny2+ny3+ny4+2 to Ny do 
    progonx(j,lamda1,ro1,c1,T); 
{                  СЛАУ в направлении оси Оy 
         в промежутке между границей x = 0 и включением 2} 
   for i:=2 to nx1 do 
    progony(i,lamda1,ro1,c1,T); 
{на левой границе включения 2} 
progonyIV(nx1+1,ny1+ny2+ny3,ny4,lamda1,0.5*(lamda1+lamda2),ro1,0.5*(
ro1+ro2),c1,0.5*(c1+c2),T); 
{во включении 2} 
   for i:=nx1+2 to nx1+nx2 do 
    progonyIV(i,ny1+ny2+ny3,ny4,lamda1,lamda2,ro1,ro2,c1,c2,T); 
{на правой границе включения 2} 
progonyIV(nx1+nx2+1,ny1+ny2+ny3,ny4,lamda1,0.5*(lamda1+lamda2),ro1,
0.5*(ro1+ro2),c1,0.5*(c1+c2),T); 
{в промежутке между включением 2 и включением 3} 
   for i:=nx1+nx2+2 to nx1+nx2+nx3 do 
    progony(i,lamda1,ro1,c1,T); 
{на левой границе включения 3} 
progonyIV(nx1+nx2+nx3+1,ny1,ny2,lamda1,0.5*(lamda1+lamda3),ro1,0.5*(
ro1+ro3),c1,0.5*(c1+c3),T); 
{во включении 3} 
   for i:=nx1+nx2+nx3+2 to nx1+nx2+nx3+nx4 do 
    progonyIV(i,ny1,ny2,lamda1,lamda3,ro1,ro3,c1,c3,T); 
{на правой границе включения 3} 
progonyIV(nx1+nx2+nx3+nx4+1,ny1,ny2,lamda1,0.5*(lamda1+lamda3),ro1,
0.5*(ro1+ro3),c1,0.5*(c1+c3),T); 
{в промежутке между включением 3 и границей х = L} 
   for i:=nx1+nx2+nx3+nx4+2 to Nx-1 do 
    progony(i,lamda1,ro1,c1,T); 
  end; {цикл с предусловием окончен} 

{выводим результат в файл} 
 Assign(g,'tempr.txt'); 
 Rewrite(g); 
 for i:=1 to Nx do 
  for j:=1 to Ny do 
   writeln(g,'  ',hx*(i-1):10:8,'  ',hy*(j-1):10:8,'   ',T[i,j]:8:5); 
 close(g); 
end. 
 
 Получено следующее распределение температуры: 
 

 
Рис. 21. Изотермы (ºС) в пластине при t = 600 c 

 
3. НЕЛИНЕЙНЫЕ ЗАДАЧИ ТЕПЛОПРОВОДНОСТИ 

3.1. ОДНОМЕРНОЕ УРАВНЕНИЕ ТЕПЛОПРОВОДНОСТИ С 
ЗАВИСЯЩИМ ОТ ТЕМПЕРАТУРЫ КОЭФФИЦИЕНТОМ 

ТЕПЛОПРОВОДНОСТИ 
 
 Рассмотренные выше решения для температурных полей в 
одномерной и двумерной постановке получены при условии 
постоянства коэффициента теплопроводности. В действительности 
коэффициент теплопроводности достаточно часто зависит от 
температуры. Например, для расчета коэффициента теплопроводности 
диоксида урана UO2 используется следующая зависимость: 

( ) 31010942.0
560

5500 T
T

T −⋅+
+

=λ .   (37) 

 Эта формула эмпирическая, коэффициенты размерные 
( )КмВт ,  К; , ⋅λT . 

 Нелинейное одномерное уравнение теплопроводности в этом 
случае будет иметь вид: 

( ) Lx
x
TT

xt
Tc <<⎟

⎠
⎞

⎜
⎝
⎛

∂
∂

λ
∂
∂

=
∂
∂

ρ 0  , .   (38) 

 Для уравнения (38) рассмотрим краевую задачу: 

 Для того чтобы дать полное математическое описание 
рассматриваемой задачи, необходимо еще задать физические условия 
однозначности. Если пластина изготовлена из диоксида урана, то           
ρ = 10950 кг/м3, с = 236 Дж/(кг⋅К). Пластина с размером м 5.0=L . На 
границах поддерживаются постоянные температуры К 373=hT  при 

0=x  и К 363=cT  при Lx = . Начальная температура области решения 
К 3230 =T . 

 Эту задачу также будем решать на равномерной сетке. 
Аналогично пункту 2.1 пластину разбиваем по толщине на N–1 равных 
промежутков. 
 Далее заменим дифференциальные операторы в (38), (39) на их 
конечно-разностные аналоги. Поскольку отличие заключается в 
появлении коэффициента теплопроводности, зависящего от 
температуры, то основной акцент сделаем на аппроксимации 
диффузионного члена. Рассмотрим сначала явно-неявную схему.  

 Таким образом, в результате аппроксимации частных 
производных соответствующими конечными разностями получаем 
следующую систему линейных алгебраических уравнений: 

получим замкнутую разностную задачу. 
 Полученную систему можно свести к наиболее общему виду: 

 Прогоночные коэффициенты находятся по формулам (8). Далее 
неизвестное поле температуры определяется по выражению (7). 
 Блок-схема аналогична представленной в пункте 2.1. 
 

Ниже приведен листинг программы для решения рассматриваемой 
задачи (на языке программирования Pascal) 

uses crt; 
const mf=500; 
type 
     vector=array[1..mf] of real; 
var {раздел описания переменных, которые мы будем использовать в 
программе} 
 i, j, N                                      : integer; 
 T, alfa, beta                            : vector; 
 ai, bi, ci, fi                              : real; 
 ro, c, h, tau                             : real; 
 Th, T0, Tc, L, t_end, time     : real; 
 f, g                                         : text; 

function lamda(x:real):real; 
{функция вычисления коэффициента теплопроводности по формуле 
(31)} 
begin 
 lamda:=5500/(560+x)+0.942*(1e-10)*x*sqr(x); 
end; 
begin 
 clrscr; 
{с клавиатуры вводим все необходимые входные параметры} 
 Writeln('Введите количество пространственных узлов, N'); 
 Readln(N); 
 Writeln('Введите окончание по времени, t_end'); 
 Readln(t_end); 
 Writeln('Введите толщину пластины, L'); 
 Readln(L); 
 Writeln('Введите плотность материала пластины, ro'); 
 Readln(ro); 
 Writeln('Введите теплоемкость материала пластины, c'); 
 Readln(c); 
 Writeln('Введите начальную температуру в К, T0'); 
 Readln(T0); 
 Writeln('Введите температуру в К на границе х=0, Th'); 
 Readln(Th); 
 Writeln('Введите температуру в К на границе х=L, Tc'); 
 Readln(Tc); 
{определяем расчетный шаг сетки по пространственной координате} 
 h:=L/(N-1); 
{определяем расчетный шаг сетки по времени} 
 tau:=t_end/100.0; 
{определяем поле температуры в начальный момент времени} 
 for i:= 1 to N do 
  T[i]:=T0; 
{проводим интегрирование нестационарного уравнения 
теплопроводности} 
 time:=0; 
 while time<t_end do {используем цикл с предусловием} 
  begin 
{увеличиваем переменную времени на шаг τ} 
   time:=time+tau; 

{определяем начальные прогоночные коэффициенты на основе левого 
граничного условия} 
   alfa[1]:=0.0; 
   beta[1]:=Th; 
{цикл с параметром для определения прогоночных коэффициентов по 
формуле (8)} 
   for i:= 2 to N-1 do 
    begin 
{ai, bi, ci, fi – коэффициенты канонического представления системы 
уравнений с трехдиагональной матрицей} 
     ai:=0.5*(lamda(T[i])+lamda(T[i+1]))/sqr(h); 
     ci:=0.5*(lamda(T[i-1])+lamda(T[i]))/sqr(h); 
     bi:=ai+ci+ro*c/tau; 
     fi:=-ro*c*T[i]/tau; 
{alfa[i], beta[i] – прогоночные коэффициенты} 
     alfa[i]:=ai/(bi-ci*alfa[i-1]); 
     beta[i]:=(ci*beta[i-1]-fi)/(bi-ci*alfa[i-1]); 
    end; 
{определяем значение температуры на правой границе на основе 
правого граничного условия} 
   T[N]:=Tc; 
{используя соотношение (7) определяем неизвестное поле 
температуры} 
   for i:= N-1 downto 1 do 
    T[i]:=alfa[i]*T[i+1]+beta[i]; 
  end; {цикл с предусловием окончен} 
{Выводим результат в файл} 
 Assign(f,'res.txt'); 
 Rewrite(f); 
 Writeln(f,'Толщина пластины L = ',L:6:4); 
 Writeln(f,'Число узлов по координате N = ',N); 
 Writeln(f,'Плотность материала пластины ro = ',ro:6:4); 
 Writeln(f,'Теплоемкость материала пластины с = ',c:6:4); 
 Writeln(f,'Начальная температура  T0 = ',T0:6:4); 
 Writeln(f,'Температура на границе x = 0, Th = ',Th:6:4); 
 Writeln(f,'Температура на границе x = L, Tc = ',Tc:6:4); 
 Writeln(f,'Результат получен с шагом по координате h = ',h:6:4); 
 Writeln(f,'Результат получен с шагом по времени tau = ',tau:6:4); 
 Writeln(f,'Температурное поле в момент времени t = ',t_end:6:4); 
 close(f); 
 Assign(g,'tempr.txt'); 

 Rewrite(g); 
 for i:=1 to N do 
  writeln(g,'  ',h*(i-1):6:3,'   ',T[i]-273:8:5); 
 close(g); 
end. 
 
 Получены следующие распределения температуры: 
 

 
Рис. 22. Распределения температуры по толщине пластины в различные 

моменты времени 
 

 Рассмотрим чисто неявную схему.  

 Таким образом, в результате аппроксимации частных 
производных соответствующими конечными разностями получаем 
следующую систему линейных алгебраических уравнений: 

получим замкнутую разностную задачу. 
 При этом видно, что полученная система уравнений является 
нелинейной, поэтому для решения этой системы воспользуемся 
методом простой итерации. Этот метод заключается в следующем – на 
каждом шаге по времени мы будем определять поле температуры, до 
тех пор пока оно не прекратит изменяться с изменением λ, т.е. 

где s – номер итерации. Видно, что система (42) уже является линейной 
относительно 1+s

iT , что позволяет воспользоваться методом прогонки и 
определить неизвестное поле температуры. Но при этом система (42) 
решается до тех пор пока поле температуры не перестанет отличаться от 
предыдущего приближения, т.е. в качестве условия остановки счета на 
данном временном слое можно использовать следующее соотношение: 

где ε  – точность вычислений. Когда условие (43) выполняется, то 
11 ++ = n

i TT . В качестве начального приближения можно рассматривать 
следующее: n

i

i TT ==0 . 

Блок-схема к рассматриваемой задаче имеет вид: 

Ниже приведен листинг программы для решения рассматриваемой 
задачи (на языке программирования Pascal) 

 
uses crt; 
const mf=500; 
         eps=1e-5; 
type 
     vector=array[1..mf] of real; 
var {раздел описания переменных, которые мы будем использовать в 
программе} 
 i, j, N, s                                 : integer; 
 T, Ts, Tn, alfa, beta              : vector; 
 ai, bi, ci, fi, max1, max2      : real; 
 ro, c, h, tau                           : real; 
 Th, T0, Tc, L, t_end, time    : real; 
 f, g, f1, f2                             : text; 
function lamda(x:real):real; 
{функция вычисления коэффициента теплопроводности по формуле 
(31)} 
begin 
 lamda:=5500/(560+x)+0.942*(1e-10)*x*sqr(x); 
end; 
begin 
 clrscr; 
{с клавиатуры вводим все необходимые входные параметры} 
 Writeln('Введите количество пространственных узлов, N'); 
 Readln(N); 
 Writeln('Введите окончание по времени, t_end'); 
 Readln(t_end); 
 Writeln('Введите толщину пластины, L'); 
 Readln(L); 
 Writeln('Введите плотность материала пластины, ro'); 
 Readln(ro); 
 Writeln('Введите теплоемкость материала пластины, c'); 
 Readln(c); 
 Writeln('Введите начальную температуру в К, T0'); 
 Readln(T0); 
 Writeln('Введите температуру в К на границе х=0, Th'); 
 Readln(Th); 
 Writeln('Введите температуру в К на границе х=L, Tc'); 
 Readln(Tc); 

{определяем расчетный шаг сетки по пространственной координате} 
 h:=L/(N-1); 
{определяем расчетный шаг сетки по времени} 
 tau:=t_end/100.0; 
{определяем поле температуры в начальный момент времени} 
 for i:= 2 to N-1 do 
  T[i]:=T0; 
{заводим файл содержащий количество итераций на каждом шаге по 
времени} 
 Assign(f1,'iter.txt'); 
 Rewrite(f1); 
{проводим интегрирование нестационарного уравнения 
теплопроводности} 
 time:=0; 
 while time<t_end do {используем цикл с предусловием} 
  begin 
{увеличиваем переменную времени на шаг τ} 
   time:=time+tau; 
{запоминаем поле температуры на предыдущем временном слое} 
   for i:= 1 to N do 
    Tn[i]:=T[i]; 
   s:=0; {будем считать число итераций} 
{цикл с постусловием, позволяющий итерационно вычислять поле 
температуры} 
   repeat 
    inc(s); 
{запоминаем поле температуры на предыдущей итерации} 
    for i:= 1 to N do 
     Ts[i]:=T[i]; 
{определяем начальные прогоночные коэффициенты на основе левого 
граничного условия} 
    alfa[1]:=0.0; 
    beta[1]:=Th; 
{цикл с параметром для определения прогоночных коэффициентов по 
формуле (8)} 
    for i:= 2 to N-1 do 
     begin 
{ai, bi, ci, fi – коэффициенты канонического представления системы 
уравнений с трехдиагональной матрицей} 
      ai:=0.5*(lamda(T[i])+lamda(T[i+1]))/sqr(h); 
      ci:=0.5*(lamda(T[i-1])+lamda(T[i]))/sqr(h); 

      bi:=ai+ci+ro*c/tau; 
      fi:=-ro*c*Tn[i]/tau; 
{alfa[i], beta[i] – прогоночные коэффициенты} 
      alfa[i]:=ai/(bi-ci*alfa[i-1]); 
      beta[i]:=(ci*beta[i-1]-fi)/(bi-ci*alfa[i-1]); 
     end; 
{определяем значение температуры на правой границе на основе 
правого граничного условия} 
    T[N]:=Tc; 
{используя соотношение (7) определяем неизвестное поле 
температуры} 
    for i:= N-1 downto 1 do 
     T[i]:=alfa[i]*T[i+1]+beta[i]; 
{определяем максимум модуля разности температур на данной и 
предыдущей итерации} 
    max1:=abs(T[1]-Ts[1]); 
    for i:= 2 to N do 
     if max1 < abs(T[i]-Ts[i]) then max1:=abs(T[i]-Ts[i]); 
{определяем максимальное по модулю значение температуры на данной 
итерации} 
    max2:=abs(T[1]); 
    for i:= 2 to N do 
     if max2 < abs(T[i]) then max2:=abs(T[i]); 
   until max1/max2<=eps; {цикл с постусловием окончен} 
   Writeln(f1,'В момент времени  ',time:6:4,'  проведено ',s,' итераций'); 
  end; {цикл с предусловием окончен} 
{выводим результат в файл} 
 Assign(f,'res.txt'); 
 Rewrite(f); 
 Writeln(f,'Толщина пластины L = ',L:6:4); 
 Writeln(f,'Число узлов по координате N = ',N); 
 Writeln(f,'Плотность материала пластины ro = ',ro:6:4); 
 Writeln(f,'Теплоемкость материала пластины с = ',c:6:4); 
 Writeln(f,'Начальная температура  T0 = ',T0:6:4); 
 Writeln(f,'Температура на границе x = 0, Th = ',Th:6:4); 
 Writeln(f,'Температура на границе x = L, Tc = ',Tc:6:4); 
 Writeln(f,'Результат получен с шагом по координате h = ',h:6:4); 
 Writeln(f,'Результат получен с шагом по времени tau = ',tau:6:4); 
 Writeln(f,'Температурное поле в момент времени t = ',t_end:6:4); 
 close(f); 
 Assign(g,'tempr.txt'); 

 Rewrite(g); 
 for i:=1 to N do 
  writeln(g,'  ',h*(i-1):6:3,'   ',T[i]-273:8:5); 
 close(g); 
 close(f1); 
end. 
 Результаты, полученные на основе чисто неявной схемы, 
полностью совпадают с результатами, полученными по явно-неявной 
схеме, число итераций при этом не превышает 3. 
 

3.2. ОДНОМЕРНОЕ УРАВНЕНИЕ ТЕПЛОПРОВОДНОСТИ С 
НЕЛИНЕЙНЫМИ ГРАНИЧНЫМИ УСЛОВИЯМИ (ИЗЛУЧЕНИЕ 

НА ГРАНИЦЕ) 
 
 В качестве примера рассмотрим теплоперенос в бесконечной 
пластине. Тем самым пренебрегаются два направления переноса тепла, 
и анализируется одномерное уравнение теплопроводности. На границах 
области решения будет моделироваться теплообмен за счет конвекции и 
излучения. Теплоперенос излучением будем рассматривать на основе 
закона Стефана-Больцмана. Таким образом, сформулированная 
физическая постановка математически будет выглядеть так: 

где 21  , εε  – приведенная степень черноты, ( )428 КмВт10669.5 ⋅⋅=σ −  – 
постоянная Стефана-Больцмана. 
 Основной интерес в сформулированной краевой задаче 
представляют нелинейные граничные условия. 
 Проведем дискретизацию нелинейных граничных условий III рода 
с погрешностью )(hO . 
 Определим первые прогоночные коэффициенты 11  и βα  из 
соотношения 1211 β+⋅α= TT . 
Итак, из второго соотношения (45) следует: 

 Видим, что прогоночный коэффициент 1β  нелинейным образом 
зависит от температуры на левой границе. Тогда для определения поля 
температуры необходимо воспользоваться, например, методом простой 
итерации. Основная идея, которого, заключается в том, чтобы на 
каждом временном слое расчет поля температуры вести до тех пор, пока 
не будет выполняться условие, вида: ε≤−+ ~

1
1

1
ss TT , где s – номер 

итерации, ε~  – точность вычислений. 
 Правое граничное условие используют для определения 
температуры NT . 

т.к. 111 −−− β+⋅α= NNNN TT , то 

 В результате получили нелинейное уравнение (47) для 
определения температуры на правой границе. Это уравнение также 
можно решить наиболее простым методом – методом простых 
итераций. 
 Проведем дискретизацию нелинейных граничных условий (45) с 
погрешностью )( 2hO . Предположим, что на границе выполняется 
уравнение теплопроводности (44). Разложим функцию )(xТ  в ряд 
Тейлора в окрестности точки 0=x  до членов второго порядка 

Из второго соотношения (45): 

Приравнивая последние два соотношения, получим: 

Таким образом, 

 Определим NT , используя правое граничное условие. 

Таким образом, т.к. 111 −−− β+⋅α= NNNN TT  

 Определим температурное поле в бетонной пластине через 600, 
1800, 3600 и 7200 секунд. Толщина пластины м 3.0=L . Начальная 
температура С500

0 =T . Бетон имеет следующие теплофизические 

характеристики λ = 0,9 Вт/(м⋅ºC), ρ = 2000 кг/м3, с = 920 Дж/(кг⋅ºC). На 
границе 0=x  и Lx =  пластина контактирует с внешней средой 
( ( )СмВт 1000 02

1 ⋅=κ , C30 0e1 =T , 5.01 =ε  и ( )СмВт 500 02
2 ⋅=κ , 

C 70 0e1 =T , 2.02 =ε ). 
 

Блок-схема к рассматриваемой задаче имеет вид: 

Ниже приведен листинг программы для решения рассматриваемой 
задачи (на языке программирования Pascal) 

uses crt; 
const mf=500; 
      sigma=5.669e-8; 
      eps=1e-5; 
type 
     vector=array[1..mf] of real; 
var {раздел описания переменных, которые мы будем использовать в 
программе} 
 i, j, N                                 : integer; 
 T, Tn, alfa, beta                 : vector; 
 ai, bi, ci, fi, d, d1               : real; 
 lamda, ro, c, h, tau             : real; 
 kapa1, kapa2, Te1, Te2     : real; 
 eps1, eps2                          : real; 
 T0, L, t_end, time              : real; 
 f, g                                     : text; 
begin 
 clrscr; 
{с клавиатуры вводим все необходимые входные параметры} 
 Writeln('Введите количество пространственных узлов, N'); 
 Readln(N); 
 Writeln('Введите окончание по времени, t_end'); 
 Readln(t_end); 
 Writeln('Введите толщину пластины, L'); 
 Readln(L); 
 Writeln('Введите коэффициент теплопроводности материала пластины, 
lamda'); 
 Readln(lamda); 
 Writeln('Введите плотность материала пластины, ro'); 
 Readln(ro); 
 Writeln('Введите теплоемкость материала пластины, c'); 
 Readln(c); 
 Writeln('Введите коэффициент теплообмена на границе х = 0, kapa1'); 
 Readln(kapa1); 
 Writeln('Введите коэффициент теплообмена на границе х = L, kapa2'); 
 Readln(kapa2); 
 Writeln('Введите температуру внешней среды относительно границы х 
= 0, Te1'); 
 Readln(Te1); 

 Writeln('Введите температуру внешней среды относительно границы х 
= L, Te2'); 
 Readln(Te2); 
 Writeln('Введите приведенную степень черноты на границе х = 0, eps1'); 
 Readln(eps1); 
 Writeln('Введите приведенную степень черноты на границе х = L, eps2'); 
 Readln(eps2); 
 Writeln('Введите начальную температуру, T0'); 
 Readln(T0); 
{определяем расчетный шаг сетки по пространственной координате} 
 h:=L/(N-1); 
{определяем расчетный шаг сетки по времени} 
 tau:=t_end/100.0; 
{определяем поле температуры в начальный момент времени} 
 for i:= 1 to N do 
  T[i]:=T0; 
{проводим интегрирование нестационарного уравнения 
теплопроводности} 
 time:=0; 
 while time<t_end do {используем цикл с предусловием} 
  begin 
{увеличиваем переменную времени на шаг τ} 
   time:=time+tau; 
{определяем alfa начальный прогоночный коэффициент на основе левого 
граничного условия, используя соотношение (48)} 
   alfa[1]:=2.0*tau*lamda/(2.0*tau*(lamda+kapa1*h)+ro*c*sqr(h)); 
{запоминаем поле температуры на предыдущем временном слое} 
   for i:=1 to N do 
    Tn[i]:=T[i]; 
{цикл с постусловием, позволяющий итерационно вычислять поле 
температуры, вследствие наличия нелинейности в левом граничном 
условии} 
   repeat 
{определяем beta начальный прогоночный коэффициент на основе 
левого граничного условия, используя соотношение (48), при этом 
начинаем итерационный цикл по левому граничному условию} 
    d:=T[1]; 
    beta[1]:=(ro*c*sqr(h)*Tn[1]+2.0*tau*kapa1*h*Te1+2.0*tau*eps1 
                   *sigma*h*(sqr(sqr(Te1))-sqr(sqr(d))))/(2.0*tau*(lamda 
                   +kapa1*h)+ro*c*sqr(h)); 

{цикл с параметром для определения прогоночных коэффициентов по 
формуле (8)} 
    for i:= 2 to N-1 do 
     begin 
{ai, bi, ci, fi – коэффициенты канонического представления системы 
уравнений с трехдиагональной матрицей} 
      ai:=lamda/sqr(h); 
      bi:=2.0*lamda/sqr(h)+ro*c/tau; 
      ci:=lamda/sqr(h); 
      fi:=-ro*c*Tn[i]/tau; 
{alfa[i], beta[i] – прогоночные коэффициенты} 
      alfa[i]:=ai/(bi-ci*alfa[i-1]); 
      beta[i]:=(ci*beta[i-1]-fi)/(bi-ci*alfa[i-1]); 
     end; 
{цикл с постусловием, позволяющий итерационно вычислить значение 
температуры на правой границе, вследствие наличия нелинейности в 
этом граничном условии} 
    repeat 
     d1:=T[N]; 
{определяем значение температуры на правой границе на основе 
правого граничного условия, используя соотношение (49)} 
     T[N]:=(ro*c*sqr(h)*Tn[N]+2.0*tau*(lamda*beta[N-1]+kapa2*h*Te2 
                 +eps2*sigma*h*(sqr(sqr(Te2))-sqr(sqr(d1))))) 
                 /(ro*c*sqr(h)+2.0*tau*(lamda*(1-alfa[N-1])+kapa2*h)); 
    until abs(d1-T[N])<=eps; {значение температуры справа определили} 
{используя соотношение (7) определяем неизвестное поле 
температуры} 
    for i:= N-1 downto 1 do 
     T[i]:=alfa[i]*T[i+1]+beta[i]; 
   until abs(d-T[1])<=eps; {значение температуры справа определили} 
  end; {цикл с предусловием окончен} 
{выводим результат в файл} 
 Assign(f,'res.txt'); 
 Rewrite(f); 
 Writeln(f,'Толщина пластины L = ',L:6:4); 
 Writeln(f,'Число узлов по координате N = ',N); 
 Writeln(f,'Коэффициент теплопроводности материала пластины lamda = 
',lamda:6:4); 
 Writeln(f,'Плотность материала пластины ro = ',ro:6:4); 
 Writeln(f,'Теплоемкость материала пластины с = ',c:6:4); 
 Writeln(f,'Начальная температура  T0 = ',T0:6:4); 

 Writeln(f,'Коэффициент теплообмена kapa1 = ',kapa1:6:4); 
 Writeln(f,'Коэффициент теплообмена kapa2 = ',kapa2:6:4); 
 Writeln(f,'Температура внешней среды Te1 = ',Te1:6:4); 
 Writeln(f,'Температура внешней среды Te2 = ',Te2:6:4); 
 Writeln(f,'Приведенная степень черноты eps1 = ',eps1:6:4); 
 Writeln(f,'Приведенная степень черноты eps2 = ',eps2:6:4); 
 Writeln(f,'Результат получен с шагом по координате h = ',h:6:4); 
 Writeln(f,'Результат получен с шагом по времени tau = ',tau:6:4); 
 Writeln(f,'Температурное поле в момент времени t = ',t_end:6:4); 
 close(f); 
 Assign(g,'tempr.txt'); 
 Rewrite(g); 
 for i:=1 to N do 
  writeln(g,'  ',h*(i-1):10:8,'   ',T[i]:8:5); 
 close(g); 
end. 
 
 Получены следующие распределения температуры: 
 

 
Рис. 23. Распределения температуры по толщине пластины в различные 

моменты времени 

3.3. ОДНОМЕРНОЕ УРАВНЕНИЕ ТЕПЛОПРОВОДНОСТИ С 
ФАЗОВЫМ ПЕРЕХОДОМ НА ГРАНИЦЕ (ИСПАРЕНИЕ 

МАТЕРИАЛА) 
 
 Рассмотрим процесс теплопроводности в пластине с испарением 
на двух границах. Математическая постановка задачи: 

давление насыщенного пара, А – коэффициент аккомодации, 0P  – 
предэкспонент фазового перехода, испQ  – энергия активации процесса 
испарения, ( )КмольДж 31.8 ⋅=R  – универсальная газовая постоянная, 
М – молекулярный вес. В качестве *P  рассмотрим атмосферное 
давление, т.е. атм

* PP = . 
 Основной интерес в сформулированной краевой задаче 
представляют нелинейные граничные условия. 
 Проведем дискретизацию нелинейного граничного условия II рода 
с погрешностью )(hO . 
 Определим первые прогоночные коэффициенты 11  и βα  из 
соотношения 1211 β+⋅α= TT . 
Итак, из второго соотношения (51): 

 Видим, что прогоночный коэффициент 1β  нелинейным образом 
зависит от температуры на левой границе. Для определения поля 
температуры необходимо воспользоваться методом простой итерации, 
основная идея которого изложена в пункте 3.2. 
 Правое граничное условие используют для определения 
температуры NT . 

т.к. 111 −−− β+⋅α= NNNN TT , то 

 В результате получили нелинейное уравнение (53) для 
определения температуры на правой границе. Это уравнение также 
можно решить методом простой итерации. 
 Проведем дискретизацию нелинейных граничных условий (51) с 
погрешностью )( 2hO . Предположим, что на границе выполняется 
уравнение теплопроводности (50). Разложим функцию )(xТ  в ряд 
Тейлора в окрестности точки 0=x  до членов второго порядка 

Из соотношения (51): 

. 
2

exp

1
1

испатм1
1

исп
01

0

M
RT

QP
RT
QPA

q
x
T

n

nn

x
+

++

= π
⋅λ

⋅⎟
⎟
⎠

⎞
⎜
⎜
⎝

⎛
−⎟⎟

⎠

⎞
⎜⎜
⎝

⎛
−⋅⋅

+
λ

−=
∂
∂  

Приравнивая последние два соотношения, получим: 

 Определим NT , используя правое граничное условие. 

Таким образом, т.к. 111 −−− β+⋅α= NNNN TT  

; 
22

2

exp

2
1

2

1

испатм1
исп

0
1

1
1

1

n
N

n
N

n
N

n
Nn

NN
n

NN

TсhTсh
M
RT

QP
RT
QPhA

qhTT

τ⋅λ⋅
ρ

−
τ⋅λ⋅

ρ

+
π

⋅λ

⋅⎟
⎟
⎠

⎞
⎜
⎜
⎝

⎛
−⎟⎟

⎠

⎞
⎜⎜
⎝

⎛
−⋅⋅⋅

+
λ

−=β+⋅α

+

+

+
+

−
+

−

 

;
2

exp

22
1

1

испатм1
исп

0

2

1

2

1
1

M
RT

QP
RT
QPhA

TсhqhсhT

n
N

n
N

n
NNN

n
N

+

+

−−
+

π
⋅λ

⋅⎟
⎟
⎠

⎞
⎜
⎜
⎝

⎛
−⎟⎟

⎠

⎞
⎜⎜
⎝

⎛
−⋅⋅⋅

−
τ⋅λ⋅

ρ
+

λ
+β=⎟⎟

⎠

⎞
⎜⎜
⎝

⎛
τ⋅λ⋅

ρ
+α−

 

( )
( )( )

.
212

exp2

12
22

1
2

1

испатм1
исп

0

2
1

2
11

M
RTсh

QP
RT
QPAh

сh
TсhqhT

n
N

N

n
N

N

n
NNn

N +

−

+

−

−+

π
⋅ρ+α−λτ

⋅⎟
⎟
⎠

⎞
⎜
⎜
⎝

⎛
−⎟⎟

⎠

⎞
⎜⎜
⎝

⎛
−⋅⋅τ

−
ρ+α−λτ
ρ+τ+λτβ

=   (55) 

 Проанализируем процесс сублимации трифторид алюминия AlF3. 
Определим температурное поле в бетонной пластине через 60, 120, 300 
секунд. Толщина образца м3.0=L . Начальная температура K 12730 =T . 
AlF3 имеет следующие теплофизические характеристики                      
λ = 60 Вт/(м⋅K), ρ = 3070 кг/м3, с = 1260 Дж/(кг⋅K), M = 0.084 кг/моль. 

.кгДж108.3 ,мВт10 ,10 ,1.0 4
исп

265
0 ⋅==== QqPA  

Блок-схема к рассматриваемой задаче имеет вид: 

Ниже приведен листинг программы для решения рассматриваемой 
задачи (на языке программирования Pascal) 

 
uses crt; 
const mf=500; 
         R=8.31; 
        eps=1e-5; 
        Patm=101325; 
type 
     vector=array[1..mf] of real; 
var {раздел описания переменных, которые мы будем использовать в 
программе} 
 i, j, N                             : integer; 
 T, Tn, alfa, beta             : vector; 
 ai, bi, ci, fi, d, d1           : real; 
 lamda, ro, c, h, tau        : real; 
 q, A, qev, P0, M           : real; 
 T0, L, t_end, time         : real; 
 f, g                                : text; 
begin 
 clrscr; 
{с клавиатуры вводим все необходимые входные параметры} 
 Writeln('Введите количество пространственных узлов, N'); 
 Readln(N); 
 Writeln('Введите окончание по времени, t_end'); 
 Readln(t_end); 
 Writeln('Введите толщину пластины, L'); 
 Readln(L); 
 Writeln('Введите коэффициент теплопроводности материала пластины, 
lamda'); 
 Readln(lamda); 
 Writeln('Введите плотность материала пластины, ro'); 
 Readln(ro); 
 Writeln('Введите теплоемкость материала пластины, c'); 
 Readln(c); 
 Writeln('Введите коэффициент аккомодации, A'); 
 Readln(A); 
 Writeln('Введите плотность теплового потока на границе, q'); 
 Readln(q); 
 Writeln('Введите тепловой эффект фазового перехода, qev'); 
 Readln(qev); 

 Writeln('Введите молярную массу, M'); 
 Readln(M); 
 Writeln('Введите предэкспонент, P0'); 
 Readln(P0); 
 Writeln('Введите начальную температуру, T0'); 
 Readln(T0); 
{определяем расчетный шаг сетки по пространственной координате} 
 h:=L/(N-1); 
{определяем расчетный шаг сетки по времени} 
 tau:=t_end/1000.0; 
{определяем поле температуры в начальный момент времени} 
 for i:= 1 to N do 
  T[i]:=T0; 
{проводим интегрирование нестационарного уравнения 
теплопроводности} 
 time:=0; 
 while time<t_end do {используем цикл с предусловием} 
  begin 
{увеличиваем переменную времени на шаг τ} 
   time:=time+tau; 
{определяем alfa начальный прогоночный коэффициент на основе левого 
граничного условия, используя соотношение (54)} 
   alfa[1]:=2.0*tau*lamda/(2.0*tau*lamda+ro*c*sqr(h)); 
{запоминаем поле температуры на предыдущем временном слое} 
   for i:=1 to N do 
    Tn[i]:=T[i]; 
{цикл с постусловием, позволяющий итерационно вычислять поле 
температуры, вследствие наличия нелинейности в левом граничном 
условии} 
   repeat 
{определяем beta начальный прогоночный коэффициент на основе 
левого граничного условия, используя соотношение (54), при этом 
начинаем итерационный цикл по левому граничному условию} 
    d:=T[1]; 
    beta[1]:=(ro*c*sqr(h)*Tn[1]+2.0*tau*h*q)/(ro*c*sqr(h)+2.0*tau 
                   *lamda)-2.0*tau*A*h*qev*(P0*exp(-qev/(R*d))-Patm) 
                   /((2.0*tau*lamda+ro*c*sqr(h))*sqrt(2.0*pi*R*d/M)); 
{цикл с параметром для определения прогоночных коэффициентов по 
формуле (8)} 
    for i:= 2 to N-1 do 
     begin 

{ai, bi, ci, fi – коэффициенты канонического представления системы 
уравнений с трехдиагональной матрицей} 
      ai:=lamda/sqr(h); 
      bi:=2.0*lamda/sqr(h)+ro*c/tau; 
      ci:=lamda/sqr(h); 
      fi:=-ro*c*Tn[i]/tau; 
{alfa[i], beta[i] – прогоночные коэффициенты} 
      alfa[i]:=ai/(bi-ci*alfa[i-1]); 
      beta[i]:=(ci*beta[i-1]-fi)/(bi-ci*alfa[i-1]); 
     end; 
{цикл с постусловием, позволяющий итерационно вычислить значение 
температуры на правой границе, вследствие наличия нелинейности в 
этом граничном условии} 
    repeat 
     d1:=T[N]; 
{определяем значение температуры на правой границе на основе 
правого граничного условия, используя соотношение (55)} 
     T[N]:=(ro*c*sqr(h)*Tn[N]+2.0*tau*(lamda*beta[N-1]+h*q))/(ro 
                 *c*sqr(h)+2.0*tau*lamda*(1-alfa[N-1]))-2.0*tau*A*h*qev 
                 *(P0*exp(-qev/(R*d1))-Patm)/((ro*c*sqr(h)+2.0*tau*lamda 
                 *(1-alfa[N-1]))*sqrt(2.0*pi*R*d1/M)); 
    until abs(d1-T[N])<=eps; {значение температуры справа определили} 
{используя соотношение (7) определяем неизвестное поле 
температуры} 
    for i:= N-1 downto 1 do 
     T[i]:=alfa[i]*T[i+1]+beta[i]; 
   until abs(d-T[1])<=eps; {значение температуры слева определили} 
  end; {цикл с предусловием окончен} 
{выводим результат в файл} 
 Assign(f,'res.txt'); 
 Rewrite(f); 
 Writeln(f,'Толщина пластины L = ',L:6:4); 
 Writeln(f,'Число узлов по координате N = ',N); 
 Writeln(f,'Коэффициент теплопроводности материала пластины lamda = 
',lamda:6:4); 
 Writeln(f,'Плотность материала пластины ro = ',ro:6:4); 
 Writeln(f,'Теплоемкость материала пластины с = ',c:6:4); 
 Writeln(f,'Коэффициент аккомодации A = ',A:6:4); 
 Writeln(f,'Плотность теплового потока на границе q = ',q:6:4); 
 Writeln(f,'Тепловой эффект фазового перехода qev = ',qev:6:4); 
 Writeln(f,'Молярная масса M = ',M:6:4); 

 Writeln(f,'Предэкспонент P0 = ',P0:6:4); 
 Writeln(f,'Начальная температура  T0 = ',T0:6:4); 
 Writeln(f,'Результат получен с шагом по координате h = ',h:6:4); 
 Writeln(f,'Результат получен с шагом по времени tau = ',tau:6:4); 
 Writeln(f,'Температурное поле в момент времени t = ',t_end:6:4); 
 close(f); 
 Assign(g,'tempr.txt'); 
 Rewrite(g); 
 for i:=1 to N do 
  writeln(g,'  ',h*(i-1):10:8,'   ',(T[i]-273):8:5); 
 close(g); 
end. 
 
 Получены следующие распределения температуры: 
 

 
Рис. 24. Распределения температуры по толщине пластины в различные 

моменты времени 

3.4. ОДНОМЕРНОЕ УРАВНЕНИЕ ТЕПЛОПРОВОДНОСТИ С 
ХИМИЧЕСКОЙ РЕАКЦИЕЙ В МАТЕРИАЛЕ (ТЕРМИЧЕСКОЕ 

РАЗЛОЖЕНИЕ) 
 
 Рассмотрим теплоперенос в бесконечной пластине, подверженной 
термическому разложению (например, полимерный материал). На 
границах осуществляется теплообмен с окружающей средой. 
Математическая постановка сформулированной задачи: 

где химq  – тепловой эффект химической реакции, 0k  – предэкспонент 
химической реакции, Е – энергия активации химической реакции, 

( )КмольДж 31.8 ⋅=R  – универсальная газовая постоянная. 
 Для решения сформулированной краевой задачи применим метод 
конечных разностей на основе неявной четырехточечной схемы 
совместно с методом простой итерации. В результате аппроксимации 
частных производных получаем следующую систему уравнений: 

 Полученную систему можно свести к наиболее общему виду: 

 Коэффициент iF  зависит от температуры на новом временном 
слое, поэтому необходимо воспользоваться методом простой итерации. 
Основная идея, которого будет заключаться в определении поля 
температуры на каждом временном слое до тех пор, пока максимальная 
разность между локальными значениями температуры на данной и на 
предыдущей итерации не будет минимальна или: 

ε≤−+ s
i

s
ii

TT 1max , 

где ε – точность вычислений. 
 Определим температурное поле в пластине через 600, 1800 и 3600 
секунд. Толщина пластины м2.0=L . Начальная температура 

K 2980 =T . Материал пластины – полимер со следующими 
теплофизическими характеристиками λ = 0,7 Вт/(м⋅К), ρ = 1500 кг/м3,    
с = 750 Дж/(кг⋅К). На границах 0=x  и Lx =  пластина контактирует с 
окружающей средой ( ( )KмВт 40 2 ⋅=κ , K243e =T ). кгВт 103

хим =q , 
4

0 103 ⋅=k , мольДж 108 4⋅=E . 
 

Блок-схема к рассматриваемой задаче имеет вид: 

Ниже приведен листинг программы для решения рассматриваемой 
задачи (на языке программирования Pascal) 

uses crt; 
const mf=500; 
         eps=1e-5; 
         R=8.31; 
type 
     vector=array[1..mf] of real; 
var {раздел описания переменных, которые мы будем использовать в 
программе} 
 i, j, N                              : integer; 
 T, Ts, Tn, alfa, beta       : vector; 
 ai, bi, ci, fi, max            : real; 
 lamda, ro, c, h, tau         : real; 
 kapa, Te, q, k0, E          : real; 
 T0, L, t_end, time         : real; 
 f, g                                 : text; 
begin 
 clrscr; 
{с клавиатуры вводим все необходимые входные параметры} 
 Writeln('Введите количество пространственных узлов, N'); 
 Readln(N); 
 Writeln('Введите окончание по времени, t_end'); 
 Readln(t_end); 
 Writeln('Введите толщину пластины, L'); 
 Readln(L); 
 Writeln('Введите коэффициент теплопроводности материала пластины, 
lamda'); 
 Readln(lamda); 
 Writeln('Введите плотность материала пластины, ro'); 
 Readln(ro); 
 Writeln('Введите теплоемкость материала пластины, c'); 
 Readln(c); 
 Writeln('Введите коэффициент теплообмена с окружающей средой, 
kapa'); 
 Readln(kapa); 
 Writeln('Введите температуру окружающей среды, Te'); 
 Readln(Te); 
 Writeln('Введите тепловой эффект химической реакции, q'); 
 Readln(q); 
 Writeln('Введите предэкспонент, k0'); 

 Readln(k0); 
 Writeln('Введите энергию активации химической реакции, E'); 
 Readln(E); 
 Writeln('Введите начальную температуру, T0'); 
 Readln(T0); 
{определяем расчетный шаг сетки по пространственной координате} 
 h:=L/(N-1); 
{определяем расчетный шаг сетки по времени} 
 tau:=t_end/100.0; 
{определяем поле температуры в начальный момент времени} 
 for i:= 1 to N do 
  T[i]:=T0; 
{проводим интегрирование нестационарного уравнения 
теплопроводности} 
 time:=0; 
 while time<t_end do {используем цикл с предусловием} 
  begin 
{увеличиваем переменную времени на шаг τ} 
   time:=time+tau; 
{определяем начальные прогоночные коэффициенты на основе левого 
граничного условия, используя соотношения (24)} 
   alfa[1]:=2.0*tau*lamda/(2.0*tau*(lamda+kapa*h)+ro*c*sqr(h)); 
   beta[1]:=(ro*c*sqr(h)*T[1]+2.0*tau*kapa*h*Te)/(2.0*tau 
                  *(lamda+kapa*h)+ro*c*sqr(h)); 
{запоминаем поле температуры на предыдущем временном слое} 
   for i:=1 to N do 
    Tn[i]:=T[i]; 
{цикл с постусловием, позволяющий итерационно вычислять поле 
температуры, вследствие наличия нелинейности в самом уравнении 
теплопроводности (56)} 
   repeat 
{запоминаем поле температуры на предыдущей итерации} 
    for i:=1 to N do 
     Ts[i]:=T[i]; 
{цикл с параметром для определения прогоночных коэффициентов по 
формуле (8)} 
    for i:= 2 to N-1 do 
     begin 
{ai, bi, ci, fi – коэффициенты канонического представления системы 
уравнений с трехдиагональной матрицей} 
      ai:=lamda/sqr(h); 

      bi:=2.0*lamda/sqr(h)+ro*c/tau; 
      ci:=lamda/sqr(h); 
      fi:=-ro*c*Tn[i]/tau-q*k0*ro*exp(-E/(R*T[i])); 
{alfa[i], beta[i] – прогоночные коэффициенты} 
      alfa[i]:=ai/(bi-ci*alfa[i-1]); 
      beta[i]:=(ci*beta[i-1]-fi)/(bi-ci*alfa[i-1]); 
     end; 
{определяем значение температуры на правой границе на основе 
правого граничного условия, используя соотношение (25)} 
     T[N]:=(ro*c*sqr(h)*Tn[N]+2.0*tau*(lamda*beta[N-1]+kapa*h*Te)) 
                 /(ro*c*sqr(h)+2.0*tau*(lamda*(1-alfa[N-1])+kapa*h)); 
{используя соотношение (7) определяем неизвестное поле 
температуры} 
    for i:= N-1 downto 1 do 
     T[i]:=alfa[i]*T[i+1]+beta[i]; 
{определяем максимум модуля разности температур на данной и 
предыдущей итерации} 
    max:=abs(T[1]-Ts[1]); 
    for i:=2 to N do 
     if max < abs(T[i]-Ts[i]) then max:=abs(T[i]-Ts[i]); 
   until max<=eps; {поле температуры на данном временном слое 
определили} 
  end; {цикл с предусловием окончен} 
{выводим результат в файл} 
 Assign(f,'res.txt'); 
 Rewrite(f); 
 Writeln(f,'Толщина пластины L = ',L:6:4); 
 Writeln(f,'Число узлов по координате N = ',N); 
 Writeln(f,'Коэффициент теплопроводности материала пластины lamda = 
',lamda:6:4); 
 Writeln(f,'Плотность материала пластины ro = ',ro:6:4); 
 Writeln(f,'Теплоемкость материала пластины с = ',c:6:4); 
 Writeln(f,'Начальная температура  T0 = ',T0:6:4); 
 Writeln(f,'Коэффициент теплообмена kapa = ',kapa:6:4); 
 Writeln(f,'Температура внешней среды Te = ',Te:6:4); 
 Writeln(f,'Тепловой эффект химической реакции q = ',q:6:4); 
 Writeln(f,'Предэкспонент k0 = ',k0:6:4); 
 Writeln(f,'Энергия активации химической реакции E = ',E:6:4); 
 Writeln(f,'Результат получен с шагом по координате h = ',h:6:4); 
 Writeln(f,'Результат получен с шагом по времени tau = ',tau:6:4); 
 Writeln(f,'Температурное поле в момент времени t = ',t_end:6:4); 

 close(f); 
 Assign(g,'tempr.txt'); 
 Rewrite(g); 
 for i:=1 to N do 
  writeln(g,'  ',h*(i-1):10:8,'   ',(T[i]-273):8:5); 
 close(g); 
end. 
 
 Получены следующие распределения температуры: 
 

 
Рис. 25. Распределения температуры по толщине пластины в различные 

моменты времени 
 

3.5. ОДНОМЕРНОЕ УРАВНЕНИЕ ТЕПЛОПРОВОДНОСТИ С 
ПОДВИЖНОЙ ГРАНИЦЕЙ (ПРОМЕРЗАНИЕ ВЛАЖНОГО 

ГРУНТА) 
 
 Постановка задачи. Влажный грунт (рис. 26) находится в талом 
состоянии и имеет начальную постоянную температуру 0T . В 
начальный момент времени на поверхности грунта внезапно 
устанавливается некоторая температура cT , которая ниже температуры 
замерзания зT . В результате образуется промерзший слой переменной 
толщины )(tf=ξ . Нижняя подвижная граница его всегда имеет 
температуру замерзания зT . На этой границе происходит переход из 
одного агрегатного состояния в другое, на что требуется теплота 

перехода ( )кгДж ,фQ . Таким образом, верхняя граница ( )ξ=x  талой 
зоны имеет постоянную температуру замерзания, а нижняя граница 
( )Lx =  – некоторую постоянную температур грунта на большой его 
глубине. Коэффициенты переноса промерзшей и талой зон различны. 
Предполагается, что перенос тепла в грунте происходит только 
вследствие теплопроводности. 
 

 
Рис. 26. Область решения 

 
 На основе всего вышесказанного математическая постановка 
задачи примет вид: 

где ρ  – плотность грунта, ( )3мкг ; w – влажность грунта (масса влаги в 
единице массы абсолютно сухого грунта), ( )кгкг . 
 Постоянные температуры, участвующие в постановке задачи: 

0TTT зc << . 
 Сформулированную задачу будем решать методом ловли фронта в 
узел пространственной сетки. Для этого вводится равномерная 
пространственная сетка: 

 Также вводится неравномерная временная сетка: 

 Следует выбирать шаг по времени 1,,1,0  ,1 −=τ + Mnn K  таким, 
чтобы за этот временной промежуток (от nt  до 1+nt ) граница фазового 
перехода сдвинулась ровно на один шаг пространственной сетки, тогда 
можно записать: 

 Рассмотрим разностную схему в первой части грунта. Для 
дискретизации первого уравнения системы (59) воспользуемся неявной 
четырехточечной схемой. 

где * ii =  – граница фазового перехода. 
 Полученную систему можно свести к наиболее общему виду: 

 Прогоночные коэффициенты находятся по формулам (8). Далее 
неизвестное поле температуры определяется по выражению (7). 
 Рассмотрим разностную схему во второй части грунта. Для 
дискретизации второго уравнения системы (59) также воспользуемся 
неявной четырехточечной схемой. 

 Полученную систему можно свести к наиболее общему виду: 

 Прогоночные коэффициенты находятся по формулам (8). Далее 
неизвестное поле температуры определяется по выражению (7). 
 Проведем дискретизацию граничного условия при )(tx ξ= : 

 Это условие необходимо для определения шага по времени на 
каждом временном слое. 
 Проведем дискретизацию с погрешностью )(hO . 

 Далее будем опускать индексы, характеризующие 
рассматриваемую часть грунта, исходя из того, что при *ii <  – часть 1, а 
при *ii >  – часть 2. 

 И следовательно, 

 Видим, что шаг по времени зависит от температуры. Тогда для 
определения поля температуры необходимо воспользоваться методом 
простой итерации. Основная идея, которого была изложена ранее. 
 Проведем дискретизацию нелинейного граничного условия (61) с 
погрешностью )( 2hO . Разложим функцию )(xТ  в ряд Тейлора в 
окрестности точки ξ=x  до членов второго порядка относительно h: 

 Используя соотношения (59) получим: 

 Аппроксимация граничного условия (61) с погрешностью )( 2hO  
примет вид: 

 Слой грунта с глубиной м3.0=L . Теплофизические 
характеристики промерзшей зоны грунта – λ1 = 2.3 Вт/(м⋅K),                  
ρ1 = 917 кг/м3, с1 = 2090 Дж/(кг⋅K)). Теплофизические характеристики 
талой зоны грунта – λ2 = 0.6 Вт/(м⋅K), ρ2 = 1000 кг/м3,                      
с2 = 4220 Дж/(кг⋅K)). Характерные температуры K 2930 =T , K 273=зT , 

K 268=сT . Теплота фазового перехода кгДж 1032.3 5
ф ⋅=Q . 

Влажность грунта кгкг 1=w . 

Блок-схема к рассматриваемой задаче имеет вид: 
 

 



 135
 



 136

Ниже приведен листинг программы для решения рассматриваемой 
задачи (на языке программирования Pascal) 

 
uses crt; 
const mf=500; 
      eps=1e-5; 
type 
     vector=array[1..mf] of real; 
var {раздел описания переменных, которые мы будем использовать в 
программе} 
 i, j, k, N                           : integer; 
 T, Tn, Ts, alfa, beta        : vector; 
 ai, bi, ci, fi, max             : real; 
 a1, lamda1, ro1, c1         : real; 
 a2, lamda2, ro2, c2         : real; 
 T0, Tfr, Tc, h, tau           : real; 
 Qfr, w, L, t_end, time     : real; 
 f, f1, g                             : text; 
begin 
 clrscr; 
{с клавиатуры вводим все необходимые входные параметры} 
 Writeln('Введите количество пространственных узлов, N'); 
 Readln(N); 
 Writeln('Введите окончание по времени, t_end'); 
 Readln(t_end); 
 Writeln('Введите глубину грунта, L'); 
 Readln(L); 
 Writeln('Введите коэффициент теплопроводности промерзшей зоны 
грунта, lamda1'); 
 Readln(lamda1); 
 Writeln('Введите плотность промерзшей зоны грунта, ro1'); 
 Readln(ro1); 
 Writeln('Введите теплоемкость промерзшей зоны грунта, c1'); 
 Readln(c1); 
 Writeln('Введите коэффициент теплопроводности талой зоны грунта, 
lamda2'); 
 Readln(lamda2); 
 Writeln('Введите плотность талой зоны грунта, ro2'); 
 Readln(ro2); 
 Writeln('Введите теплоемкость талой зоны грунта, c2'); 
 Readln(c2); 

 Writeln('Введите начальную температуру, T0'); 
 Readln(T0); 
 Writeln('Введите температуру на границе х = 0, Tc'); 
 Readln(Tc); 
 Writeln('Введите температуру фазового перехода, Tfr'); 
 Readln(Tfr); 
 Writeln('Введите теплоту фазового перехода, Qfr'); 
 Readln(Qfr); 
 Writeln('Введите влагосодержание грунта, w'); 
 Readln(w); 
{определяем расчетный шаг сетки по пространственной координате} 
 h:=L/(N-1); 
{определяем коэффициенты температуропроводности} 
 a1:=lamda1/(ro1*c1); 
 a2:=lamda2/(ro2*c2); 
{определяем поле температуры в начальный момент времени} 
 time:=0; 
 for i:= 1 to N do 
  T[i]:=T0; 
{определяем файл, содержащий шаги по времени} 
 Assign(f1,'time.txt'); 
 Rewrite(f1); 
{определяем положение границы фазового перехода} 
 k:=1; 
{проводим интегрирование нестационарного уравнения 
теплопроводности} 
 while time<t_end do {используем цикл с предусловием} 
  begin 
{запоминаем поле температуры на предыдущем временном слое} 
   for i:= 1 to N do 
    Tn[i]:=T[i]; 
{граница фазового перехода на каждом соответствующем временном 
шаге смещается на пространственный шаг вправо} 
   inc(k); 
{цикл с постусловием, позволяющий итерационно вычислять поле 
температуры, вследствие наличия нелинейности в граничном условии 
(61)} 
   repeat 
{запоминаем поле температуры на предыдущей итерации} 
    for i:= 1 to N do 
     Ts[i]:=T[i]; 

{определяем соответствующий шаг по времени} 
    tau:=(2.0*a1*a2*Qfr*w*0.5*(ro1+ro2)*sqr(h)-sqr(h)*(lamda1*a2+lamda2 
             *a1)*(Tfr-Tn[k]))/(2.0*a1*a2*(lamda1*(Tfr-T[k-1])-lamda2 
             *(T[k+1]-Tfr))); 
{определяем начальные прогоночные коэффициенты на основе левого 
граничного условия} 
    alfa[1]:=0.0; 
    beta[1]:=Tc; 
{цикл с параметром для определения прогоночных коэффициентов по 
формуле (8)} 
    for i:= 2 to k-1 do 
     begin 
{ai, bi, ci, fi – коэффициенты канонического представления системы 
уравнений с трехдиагональной матрицей} 
      ai:=a1/sqr(h); 
      bi:=2.0*a1/sqr(h)+1.0/tau; 
      ci:=a1/sqr(h); 
      fi:=-Tn[i]/tau; 
{alfa[i], beta[i] – прогоночные коэффициенты} 
      alfa[i]:=ai/(bi-ci*alfa[i-1]); 
      beta[i]:=(ci*beta[i-1]-fi)/(bi-ci*alfa[i-1]); 
     end; 
{определяем значение температуры на границе фазового перехода} 
    T[k]:=Tfr; 
{используя соотношение (7) определяем неизвестное поле температуры 
в промерзшей зоне грунта} 
    for i:= k-1 downto 1 do 
     T[i]:=alfa[i]*T[i+1]+beta[i]; 
{определяем начальные прогоночные коэффициенты для талой зоны 
грунта на основе условия на границе раздела двух сред} 
    alfa[k]:=0.0; 
    beta[k]:=Tfr; 
{цикл с параметром для определения прогоночных коэффициентов по 
формуле (8)} 
    for i:= k+1 to N-1 do 
     begin 
{ai, bi, ci, fi – коэффициенты канонического представления системы 
уравнений с трехдиагональной матрицей} 
      ai:=a2/sqr(h); 
      bi:=2.0*a2/sqr(h)+1.0/tau; 
      ci:=a2/sqr(h); 

      fi:=-Tn[i]/tau; 
{alfa[i], beta[i] – прогоночные коэффициенты} 
      alfa[i]:=ai/(bi-ci*alfa[i-1]); 
      beta[i]:=(ci*beta[i-1]-fi)/(bi-ci*alfa[i-1]); 
     end; 
{определяем значение температуры на правой границе на основе 
граничного условия, используя соотношение (21)} 
    T[N]:=(2.0*a2*tau*beta[N-1]+sqr(h)*Tn[N])/(2.0*a2*tau 
               *(1.0-alfa[N-1])+sqr(h)); 
{используя соотношение (7) определяем неизвестное поле температуры 
в талой зоне грунта} 
    for i:= N-1 downto k do 
     T[i]:=alfa[i]*T[i+1]+beta[i]; 
{определяем максимум модуля разности между соответствующими 
значениями температуры на текущей и предыдущей итерации} 
    max:=abs(T[1]-Ts[1]); 
    for i:= 2 to N do 
     if max < abs(T[i]-Ts[i]) then max:=abs(T[i]-Ts[i]); 
   until max<=eps; {поле температуры на данном временном слое 
определили} 
{увеличиваем переменную времени на шаг τn+1} 
   time:=time+tau; 
{формируем файл, содержащий шаги по времени τn+1} 
   Writeln(f1,'tau',(k-1),' = ',tau:6:4); 
  end; 
{выводим результат в файл} 
 Assign(f,'res.txt'); 
 Rewrite(f); 
 Writeln(f,'Глубина грунта L = ',L:6:4); 
 Writeln(f,'Число узлов по координате N = ',N); 
 Writeln(f,'Коэффициент теплопроводности промерзшей зоны грунта 
lamda1 = ',lamda1:6:4); 
 Writeln(f,'Плотность промерзшей зоны грунта ro1 = ',ro1:6:4); 
 Writeln(f,'Теплоемкость промерзшей зоны грунта с1 = ',c1:6:4); 
 Writeln(f,'Коэффициент теплопроводности талой зоны грунта lamda2 = 
',lamda2:6:4); 
 Writeln(f,'Плотность талой зоны грунта ro2 = ',ro2:6:4); 
 Writeln(f,'Теплоемкость талой зоны грунта с2 = ',c2:6:4); 
 Writeln(f,'Начальная температура T0 = ',T0:6:4); 
 Writeln(f,'Температура на границе х = 0',Tc:6:4); 
 Writeln(f,'Температура фазового перехода',Tfr:6:4); 

 Writeln(f,'Теплота фазового перехода',Qfr:6:4); 
 Writeln(f,'Влагосодержание грунта',w:6:4); 
 Writeln(f,'Результат получен с шагом по координате h = ',h:6:4); 
 Writeln(f,'Температурное поле в момент времени t = ',time:6:4); 
 close(f); 
 close(f1); 
 Assign(g,'tempr.txt'); 
 Rewrite(g); 
 for i:=1 to N do 
  writeln(g,'  ',h*(i-1):10:8,'   ',(T[i]-273):8:5); 
 close(g); 
end. 
 
 Получены следующие распределения температуры: 
 

 
Рис. 27. Распределения температуры по глубине грунта 

 
3.6. ДВУМЕРНОЕ УРАВНЕНИЕ ТЕПЛОПРОВОДНОСТИ С 

ИЗЛУЧЕНИЕМ НА ГРАНИЦАХ 
 
 Проанализируем процесс теплопереноса в пластине, на двух 
границах которой осуществляется теплообмен с внешней средой за счет 
излучения и конвекции. Область решения имеет вид аналогичный рис. 
17. 
 Пластина с размерами м3.0== HL . Материал пластины – 
твердая резина (λ = 0.16 Вт/(м⋅ºC), ρ = 1190 кг/м3, с = 1900 Дж/(кг⋅ºC)). 

Начальная температура области решения C 30 0
0 =T . 

( )СмВт 50 02
1 ⋅=κ , C20 0e1 =T , 8.0=ε , ( )СмВт 35 02

2 ⋅=κ , 
C 35 0e2 =T . 

 Математическая постановка задачи будет иметь вид: 

 Начальные и граничные условия запишутся следующим образом: 

 Рассматриваемая задача объединяет в себе постановки 2.6 и 3.2. 
Для решения сформулированной задачи (64), (65) введем равномерную 
пространственно-временную сетку. 
 Дискретизацию уравнения (64) также как и в пункте 2.6 будем 
проводить на основе локально одномерной схемы А.А. Самарского. 
Решение полученных систем линейных алгебраических уравнений 
проводится методом прогонки, при этом необходимо учесть, что на 
границе присутствует излучение (пункт 3.2), которое моделируется 
нелинейным соотношением, поэтому необходимо воспользоваться 
методом простой итерации. 

Блок-схема к рассматриваемой задаче имеет вид: 

Ниже приведен листинг программы для решения рассматриваемой 
задачи (на языке программирования Pascal) 

 
uses crt; 
const mf=72; 
         sigma=5.669e-8; 
         eps=1e-5; 
type 
     vector1=array[1..mf] of real; 
     vector2=array[1..mf,1..mf] of real; 
var {раздел описания переменных, которые мы будем использовать в 
программе} 
 i, j, Nx, Ny                                     : integer; 
 T, Tn                                              : vector2; 
 alfa, beta                                        : vector1; 
 ai, bi, ci, fi, d, d1                           : real; 
 lamda, ro, c                                   : real; 
 kapa1, kapa2, Te1, Te2, eps1       : real; 
 hx, hy, tau, t_end, time                 : real; 
 T0, L, H                                        : real; 
 f, g                                                 : text; 
begin 
 clrscr; 
{с клавиатуры вводим все необходимые входные параметры} 
 Writeln('Введите количество пространственных узлов в пластине по оси 
х, Nx'); 
 Readln(Nx); 
 Writeln('Введите количество пространственных узлов в пластине по оси 
y, Ny'); 
 Readln(Ny); 
 Writeln('Введите окончание по времени, t_end'); 
 Readln(t_end); 
 Writeln('Введите длину пластины, L'); 
 Readln(L); 
 Writeln('Введите толщину пластины, H'); 
 Readln(H); 
 Writeln('Введите коэффициент теплопроводности материала пластины, 
lamda'); 
 Readln(lamda); 
 Writeln('Введите плотность материала пластины, ro'); 
 Readln(ro); 

 Writeln('Введите теплоемкость материала пластины, c'); 
 Readln(c); 
 Writeln('Введите коэффициент теплообмена на границах х,y = 0, kapa1'); 
 Readln(kapa1); 
 Writeln('Введите коэффициент теплообмена на границах х = L, y = H, 
kapa2'); 
 Readln(kapa2); 
 Writeln('Введите температуру внешней среды относительно границ х,y 
= 0, Te1'); 
 Readln(Te1); 
 Writeln('Введите температуру внешней среды относительно границ х = 
L, y = H, Te2'); 
 Readln(Te2); 
 Writeln('Введите приведенную степень черноты, eps1'); 
 Readln(eps1); 
 Writeln('Введите начальную температуру, T0'); 
 Readln(T0); 
{определяем расчетные шаги сетки по пространственным 
координатам} 
 hx:=L/(Nx-1); 
 hy:=H/(Ny-1); 
{определяем расчетный шаг сетки по времени} 
 tau:=t_end/1000.0; 
{определяем поле температуры в начальный момент времени} 
 for i:= 1 to Nx do 
  for j:= 1 to Ny do 
    T[i,j]:=T0; 
{проводим интегрирование нестационарного уравнения 
теплопроводности} 
 time:=0; 
 while time<t_end do {используем цикл с предусловием} 
  begin 
{увеличиваем переменную времени на шаг τ} 
   time:=time+tau; 
{запоминаем поле температуры на n-ом временном слое} 
   for i:=1 to Nx do 
    for j:=1 to Ny do 
     Tn[i,j]:=T[i,j]; 
{решаем СЛАУ в направлении оси Ох для определения поля 
температуры на промежуточном (n+1/2) временном слое} 
   for j:=1 to Ny do 

    begin 
{определяем alfa начальный прогоночный коэффициент на основе левого 
граничного условия, используя соотношение (48)} 
     alfa[1]:=2.0*tau*lamda/(2.0*tau*(lamda+kapa1*hx)+ro*c*sqr(hx)); 
{цикл с постусловием, позволяющий итерационно вычислять поле 
температуры, вследствие наличия нелинейности в левом граничном 
условии} 
     repeat 
{определяем beta начальный прогоночный коэффициент на основе 
левого граничного условия, используя соотношение (48), при этом 
начинаем итерационный цикл по левому граничному условию} 
      d:=T[1,j]; 
      beta[1]:=(ro*c*sqr(hx)*Tn[1,j]+2.0*tau*kapa1*hx*Te1+2.0*tau 
                     *eps1*sigma*hx*(sqr(sqr(Te1))-sqr(sqr(d)))) 
                     /(2.0*tau*(lamda+kapa1*hx)+ro*c*sqr(hx)); 
{цикл с параметром для определения прогоночных коэффициентов по 
формуле (8)} 
      for i:= 2 to Nx-1 do 
       begin 
{ai, bi, ci, fi – коэффициенты канонического представления СЛАУ с 
трехдиагональной матрицей} 
        ai:=lamda/sqr(hx); 
        bi:=2.0*lamda/sqr(hx)+ro*c/tau; 
        ci:=lamda/sqr(hx); 
        fi:=-ro*c*Tn[i,j]/tau; 
{alfa[i], beta[i] – прогоночные коэффициенты} 
        alfa[i]:=ai/(bi-ci*alfa[i-1]); 
        beta[i]:=(ci*beta[i-1]-fi)/(bi-ci*alfa[i-1]); 
       end; 
{определяем значение температуры на правой границе, используя 
соотношение (21) при условии, что q2 = 0} 
      T[Nx,j]:=(ro*c*sqr(hx)*Tn[Nx,j]+2.0*tau*lamda*beta[Nx-1]) 
                     /(ro*c*sqr(hx)+2.0*tau*lamda*(1-alfa[Nx-1])); 
{используя соотношение (7) определяем неизвестное поле температуры 
на промежуточном (n+1/2) временном слое} 
      for i:= Nx-1 downto 1 do 
       T[i,j]:=alfa[i]*T[i+1,j]+beta[i]; 
     until abs(d-T[1,j])<=eps; {значение температуры на левой границе 
определили} 
    end; {поле температуры на промежуточном (n+1/2) временном слое 
определили} 

{решаем СЛАУ в направлении оси Оу для определения поля 
температуры на целом (n+1) временном слое} 
   for i:=1 to Nx do 
    begin 
{определяем начальные прогоночные коэффициенты на основе нижнего 
граничного условия, используя соотношения (20) при условии, что         
q1 = 0} 
     alfa[1]:=2.0*tau*lamda/(2.0*tau*lamda+ro*c*sqr(hy)); 
     beta[1]:=ro*c*sqr(hy)*T[i,1]/(2.0*tau*lamda+ro*c*sqr(hy)); 
{цикл с параметром для определения прогоночных коэффициентов по 
формуле (8)} 
     for j:= 2 to Ny-1 do 
      begin 
{ai, bi, ci, fi – коэффициенты канонического представления СЛАУ с 
трехдиагональной матрицей} 
       ai:=lamda/sqr(hy); 
       bi:=2.0*lamda/sqr(hy)+ro*c/tau; 
       ci:=lamda/sqr(hy); 
       fi:=-ro*c*T[i,j]/tau; 
{alfa[j], beta[j] – прогоночные коэффициенты} 
       alfa[j]:=ai/(bi-ci*alfa[j-1]); 
       beta[j]:=(ci*beta[j-1]-fi)/(bi-ci*alfa[j-1]); 
      end; 
{запоминаем значение температуры на правой границе с 
промежуточного (n+1/2) временного слоя} 
     d:=T[i,Ny]; 
{цикл с постусловием, позволяющий итерационно вычислить значение 
температуры на правой границе, вследствие наличия нелинейности в 
этом граничном условии} 
     repeat 
      d1:=T[i,Ny]; 
{определяем значение температуры на правой границе на основе 
правого граничного условия, используя соотношение (49)} 
      T[i,Ny]:=(ro*c*sqr(hy)*d+2.0*tau*(lamda*beta[Ny-1]+kapa2*hy*Te2 
                     +eps1*sigma*hy*(sqr(sqr(Te2))-sqr(sqr(d1)))))/(ro*c*sqr(hy) 
                     +2.0*tau*(lamda*(1-alfa[Ny-1])+kapa2*hy)); 
     until abs(d1-T[i,Ny])<=eps; {значение температуры на правой 
границе определили} 
{используя соотношение (7) определяем неизвестное поле температуры 
на целом (n+1) временном слое} 
     for j:= Ny-1 downto 1 do 

      T[i,j]:=alfa[j]*T[i,j+1]+beta[j]; 
    end; {поле температуры на целом (n+1) временном слое определили} 
  end; {цикл с предусловием окончен} 
{выводим результат в файл} 
 Assign(f,'res.txt'); 
 Rewrite(f); 
 Writeln(f,'Длина пластины L = ',L:6:4); 
 Writeln(f,'Толщина пластины H = ',H:6:4); 
 Writeln(f,'Число узлов по пространственной координате x в пластине 
Nx = ',Nx); 
 Writeln(f,'Число узлов по пространственной координате y в пластине 
Ny = ',Ny); 
 Writeln(f,'Коэффициент теплопроводности материала пластины lamda = 
',lamda:6:4); 
 Writeln(f,'Плотность материала пластины ro = ',ro:6:4); 
 Writeln(f,'Теплоемкость материала пластины с = ',c:6:4); 
 Writeln(f,'Начальная температура T0 = ',T0:6:4); 
 Writeln(f,'Коэффициент теплообмена kapa1 = ',kapa1:6:4); 
 Writeln(f,'Коэффициент теплообмена kapa2 = ',kapa2:6:4); 
 Writeln(f,'Температура внешней среды Te1 = ',Te1:6:4); 
 Writeln(f,'Температура внешней среды Te2 = ',Te2:6:4); 
 Writeln(f,'Приведенная степень черноты eps1 = ',eps1:6:4); 
 Writeln(f,'Результат получен с шагом по координате x hx = ',hx:6:4); 
 Writeln(f,'Результат получен с шагом по координате y hy = ',hy:6:4); 
 Writeln(f,'Результат получен с шагом по времени tau = ',tau:6:4); 
 Writeln(f,'Температурное поле в момент времени t = ',t_end:6:4); 
 close(f); 
 Assign(g,'tempr.txt'); 
 Rewrite(g); 
 for i:=1 to Nx do 
  for j:=1 to Ny do 
   writeln(g,'  ',hx*(i-1):10:8,'  ',hy*(j-1):10:8,'   ',T[i,j]:8:5); 
 close(g); 
end. 

 Получено следующее распределение температуры: 
 

 
Рис. 28. Распределение температуры (ºС) в пластине при t = 36000 с 

 
3.7. ДВУМЕРНОЕ УРАВНЕНИЕ ТЕПЛОПРОВОДНОСТИ С 

ФАЗОВЫМ ПЕРЕХОДОМ НА ОДНОЙ ИЗ ГРАНИЦ 
 
 Проанализируем процесс теплопереноса в пластине, две границы 
которой являются адиабатическими, а на одной моделируется процесс 
испарения. Область решения имеет вид аналогичный рис. 17. 
 Пластина с размерами м3.0== HL . Материал пластины – AlF3  
(λ = 60 Вт/(м⋅K), ρ = 3070 кг/м3, с = 1260 Дж/(кг⋅K), M = 0.084 кг/моль). 
Начальная температура области решения K 12730 =T . 

 Математическая постановка задачи будет иметь вид: 

 Начальные и граничные условия запишутся следующим образом: 

давление насыщенного пара. 
 Рассматриваемая задача объединяет в себе постановки 2.6 и 3.3. 
Для решения сформулированной задачи (66), (67) введем равномерную 
пространственно-временную сетку. 
 Дискретизацию уравнения (66) также как и в пункте 2.6 будем 
проводить на основе локально одномерной схемы А.А. Самарского. 
Решение полученных систем линейных алгебраических уравнений 
проводится методом прогонки, при этом необходимо учесть, что на 
границе присутствует процесс испарения материала (пункт 3.3), 
который моделируется нелинейным соотношением, поэтому 
необходимо воспользоваться методом простой итерации. 

Блок-схема к рассматриваемой задаче имеет вид: 

Ниже приведен листинг программы для решения рассматриваемой 
задачи (на языке программирования Pascal) 

 
uses crt; 
const mf=102; 
         R=8.31; 
        eps=1e-5; 
        Patm=101325; 
type 
     vector1=array[1..mf] of real; 
     vector2=array[1..mf,1..mf] of real; 
var {раздел описания переменных, которые мы будем использовать в 
программе} 
 i, j, Nx, Ny                      : integer; 
 T                                     : vector2; 
 alfa, beta                         : vector1; 
 ai, bi, ci, fi, d, d1             : real; 
 lamda, ro, c                     : real; 
 q, A, qev, P0, M              : real; 
 hx, hy, tau, t_end, time    : real; 
 T0, Te, kapa, L, H           : real; 
 f, g                                   : text; 
begin 
 clrscr; 
{с клавиатуры вводим все необходимые входные параметры} 
 Writeln('Введите количество пространственных узлов в пластине по оси 
х, Nx'); 
 Readln(Nx); 
 Writeln('Введите количество пространственных узлов в пластине по оси 
y, Ny'); 
 Readln(Ny); 
 Writeln('Введите окончание по времени, t_end'); 
 Readln(t_end); 
 Writeln('Введите длину пластины, L'); 
 Readln(L); 
 Writeln('Введите толщину пластины, H'); 
 Readln(H); 
 Writeln('Введите коэффициент теплопроводности материала пластины, 
lamda'); 
 Readln(lamda); 
 Writeln('Введите плотность материала пластины, ro'); 

 Readln(ro); 
 Writeln('Введите теплоемкость материала пластины, c'); 
 Readln(c); 
 Writeln('Введите коэффициент аккомодации, A'); 
 Readln(A); 
 Writeln('Введите плотность теплового потока на границе, q'); 
 Readln(q); 
 Writeln('Введите тепловой эффект фазового перехода, qev'); 
 Readln(qev); 
 Writeln('Введите молярную массу, M'); 
 Readln(M); 
 Writeln('Введите предэкспонент, P0'); 
 Readln(P0); 
 Writeln('Введите коэффициент теплообмена на границе y = 0, kapa'); 
 Readln(kapa); 
 Writeln('Введите температуру внешней среды относительно границы y 
= 0, Te'); 
 Readln(Te); 
 Writeln('Введите начальную температуру, T0'); 
 Readln(T0); 
{определяем расчетные шаги сетки по пространственным 
координатам} 
 hx:=L/(Nx-1); 
 hy:=H/(Ny-1); 
{определяем расчетный шаг сетки по времени} 
 tau:=t_end/1000.0; 
{определяем поле температуры в начальный момент времени} 
 for i:= 1 to Nx do 
  for j:= 1 to Ny do 
    T[i,j]:=T0; 
{проводим интегрирование нестационарного уравнения 
теплопроводности} 
 time:=0; 
 while time<t_end do {используем цикл с предусловием} 
  begin 
{увеличиваем переменную времени на шаг τ} 
   time:=time+tau; 
{решаем СЛАУ в направлении оси Ох для определения поля 
температуры на промежуточном (n+1/2) временном слое} 
   for j:=1 to Ny do 
    begin 

{определяем начальные прогоночные коэффициенты на основе левого 
граничного условия, используя соотношение (20) при условии, что         
q1 = 0} 
     alfa[1]:=2.0*tau*lamda/(2.0*tau*lamda+ro*c*sqr(hx)); 
     beta[1]:=ro*c*sqr(hx)*T[1,j]/(2.0*tau*lamda+ro*c*sqr(hx)); 
{цикл с параметром для определения прогоночных коэффициентов по 
формуле (8)} 
     for i:= 2 to Nx-1 do 
       begin 
{ai, bi, ci, fi – коэффициенты канонического представления СЛАУ с 
трехдиагональной матрицей} 
        ai:=lamda/sqr(hx); 
        bi:=2.0*lamda/sqr(hx)+ro*c/tau; 
        ci:=lamda/sqr(hx); 
        fi:=-ro*c*T[i,j]/tau; 
{alfa[i], beta[i] – прогоночные коэффициенты} 
        alfa[i]:=ai/(bi-ci*alfa[i-1]); 
        beta[i]:=(ci*beta[i-1]-fi)/(bi-ci*alfa[i-1]); 
       end; 
{определяем значение температуры на правой границе, используя 
соотношение (21) при условии, что q2 = 0} 
     T[Nx,j]:=(ro*c*sqr(hx)*T[Nx,j]+2.0*tau*lamda*beta[Nx-1]) 
                    /(ro*c*sqr(hx)+2.0*tau*lamda*(1-alfa[Nx-1])); 
{используя соотношение (7) определяем неизвестное поле температуры 
на промежуточном (n+1/2) временном слое} 
     for i:= Nx-1 downto 1 do 
       T[i,j]:=alfa[i]*T[i+1,j]+beta[i]; 
    end; {поле температуры на промежуточном (n+1/2) временном слое 
определили} 
{решаем СЛАУ в направлении оси Оу для определения поля 
температуры на целом (n+1) временном слое} 
   for i:=1 to Nx do 
    begin 
{определяем начальные прогоночные коэффициенты на основе нижнего 
граничного условия, используя соотношения (24)} 
     alfa[1]:=2.0*tau*lamda/(2.0*tau*(lamda+kapa*hy)+ro*c*sqr(hy)); 
     beta[1]:=(ro*c*sqr(hy)*T[i,1]+2.0*tau*kapa*hy*Te) 
                    /(2.0*tau*(lamda+kapa*hy)+ro*c*sqr(hy)); 
{цикл с параметром для определения прогоночных коэффициентов по 
формуле (8)} 
     for j:= 2 to Ny-1 do 

      begin 
{ai, bi, ci, fi – коэффициенты канонического представления СЛАУ с 
трехдиагональной матрицей} 
       ai:=lamda/sqr(hy); 
       bi:=2.0*lamda/sqr(hy)+ro*c/tau; 
       ci:=lamda/sqr(hy); 
       fi:=-ro*c*T[i,j]/tau; 
{alfa[j], beta[j] – прогоночные коэффициенты} 
       alfa[j]:=ai/(bi-ci*alfa[j-1]); 
       beta[j]:=(ci*beta[j-1]-fi)/(bi-ci*alfa[j-1]); 
      end; 
{запоминаем значение температуры на верхней границе с 
промежуточного (n+1/2) временного слоя} 
     d:=T[i,Ny]; 
{цикл с постусловием, позволяющий итерационно вычислить значение 
температуры на верхней границе, вследствие наличия нелинейности в 
этом граничном условии} 
     repeat 
      d1:=T[i,Ny]; 
{определяем значение температуры на верхней границе на основе 
верхнего граничного условия, используя соотношение (55)} 
      T[i,Ny]:=(ro*c*sqr(hy)*d+2.0*tau*(lamda*beta[Ny-1]+h*q)) 
                     /(ro*c*sqr(hy)+2.0*tau*lamda*(1-alfa[Ny-1]))-2.0*tau 
                    *A*hy*qev*(P0*exp(-qev/(R*d1))-Patm)/((ro*c*sqr(hy) 
                    +2.0*tau*lamda*(1-alfa[Ny-1]))*sqrt(2.0*pi*R*d1/M)); 
     until abs(d1-T[i,Ny])<=eps; {значение температуры на верхней 
границе определили} 
{используя соотношение (7) определяем неизвестное поле температуры 
на целом (n+1) временном слое} 
     for j:= Ny-1 downto 1 do 
      T[i,j]:=alfa[j]*T[i,j+1]+beta[j]; 
    end; {поле температуры на целом (n+1) временном слое определили} 
  end; {цикл с предусловием окончен} 
{выводим результат в файл} 
 Assign(f,'res.txt'); 
 Rewrite(f); 
 Writeln(f,'Длина пластины L = ',L:6:4); 
 Writeln(f,'Толщина пластины H = ',H:6:4); 
 Writeln(f,'Число узлов по пространственной координате x в пластине 
Nx = ',Nx); 

 Writeln(f,'Число узлов по пространственной координате y в пластине 
Ny = ',Ny); 
 Writeln(f,'Коэффициент теплопроводности материала пластины lamda = 
',lamda:6:4); 
 Writeln(f,'Плотность материала пластины ro = ',ro:6:4); 
 Writeln(f,'Теплоемкость материала пластины с = ',c:6:4); 
 Writeln(f,'Начальная температура T0 = ',T0:6:4); 
 Writeln(f,'Коэффициент теплообмена kapa = ',kapa:6:4); 
 Writeln(f,'Температура внешней среды Te = ',Te:6:4); 
 Writeln(f,'Коэффициент аккомодации A = ',A:6:4); 
 Writeln(f,'Плотность теплового потока на границе q = ',q:6:4); 
 Writeln(f,'Тепловой эффект фазового перехода qev = ',qev:6:4); 
 Writeln(f,'Молярная масса M = ',M:6:4); 
 Writeln(f,'Предэкспонент P0 = ',P0:6:4); 
 Writeln(f,'Результат получен с шагом по координате x hx = ',hx:6:4); 
 Writeln(f,'Результат получен с шагом по координате y hy = ',hy:6:4); 
 Writeln(f,'Результат получен с шагом по времени tau = ',tau:6:4); 
 Writeln(f,'Температурное поле в момент времени t = ',t_end:6:4); 
 close(f); 
 Assign(g,'tempr.txt'); 
 Rewrite(g); 
 for i:=1 to Nx do 
  for j:=1 to Ny do 
   writeln(g,'  ',hx*(i-1):10:8,'  ',hy*(j-1):10:8,'   ',(T[i,j]-273):8:5); 
 close(g); 
end. 

 Получено следующее распределение температуры: 
 

 
Рис. 29. Распределение температуры (ºС) в пластине при t = 120 с 

 
3.8. КВАЗИТРЕХМЕРНАЯ МОДЕЛЬ 

(ИЗЛУЧЕНИЕ И КОНВЕКЦИЯ ПО ТРЕТЬЕЙ КООРДИНАТЕ) 
 
 Проанализируем теплоперенос в пластине толщины h (рис. 30). 
Учтем перенос теплоты конвекцией и излучением с двух сторон 
пластины (по третьей координате). При этом уравнение 
теплопроводности является двумерным. Такой подход при решении 
многих пространственных задач позволяет значительно сократить время 
счета, но при этом получить достаточно точные результаты. 
 

 
Рис. 30. Область решения 

 Пластина с размерами м 1.0  м, 3.0 === hLL yx . Материал 
пластины – дерево (дуб) (λ = 0.3 Вт/(м⋅ºC), ρ = 800 кг/м3,                      
с = 2400 Дж/(кг⋅ºC)). Начальная температура области решения 

C 25 0

C 10  C, 50 00 == ch TT . 
 Математическая постановка задачи будет иметь вид: 

 Начальные и граничные условия запишутся следующим образом: 

 Дискретизацию уравнения (68) будем проводить на основе 
локально одномерной схемы А.А. Самарского [2]. 
В результате получим: 

 Разностные уравнения (70), (71) сводятся к стандартному 
трехдиагональному виду и решаются последовательно методом 
прогонки (пункт 2.1). Сначала для всей области решается уравнение 

(70), после того как его решение будет найдено, переходят к решению 
уравнения (71). 
 Рассмотрим решение уравнения (70) методом прогонки. Приведем 

это уравнение к виду i

n

jii

n

jii

n

jii FTCTBTA =+−
+

−

++

+
2
1

,1
2
1

,
2
1

,1 . Тогда 
коэффициенты iii CBA  , ,  примут вид: 

( )
.

    ,2    ,

4

2
1

,
4e

e
,

22

h

TT

h
TTc

Fc
hh

B
h

CA

n

ji

n
ji

i
x

i
x

ii

⎟
⎟
⎠

⎞
⎜
⎜
⎝

⎛
−

εσ−

κ
−

τ
⋅⋅ρ

−=
τ
⋅ρ

+
κ

+
λ⋅

=
λ

==

+  

 Для определения прогоночных коэффициентов по соотношению 
(8) необходимо найти 11  и βα  из левого граничного условия. Далее 

определяя значение 2
1

,

+n

jNx
T  из правого граничного условия, находят поле 

температуры 2
1

,

+n

jiT  на промежуточном временном слое по формулам (7). 

При этом, поскольку iF  является функцией 2
1

,

+n

jiT , то необходимо 
воспользоваться методом простой итерации, аналогично пункту 3.4. 
После этого приступают к решению уравнения (71). Этапы решения 
уравнения (71) аналогичны решению уравнения (70). А именно, 
приведем уравнение (71) к виду j

 Для определения прогоночных коэффициентов по соотношению 
(8) необходимо найти 11  и βα  из левого граничного условия. Далее 
определяя значение 1

,
+n
Ni y

T  из правого граничного условия, находят поле 

температуры 1
,
+n
jiT  на целом временном слое по формулам (7). При этом, 



 162

поскольку iF  является функцией 1
,
+n
jiT , то также необходимо 

воспользоваться методом простой итерации. 
 

Блок-схема к рассматриваемой задаче имеет вид: 

Ниже приведен листинг программы для решения рассматриваемой 
задачи (на языке программирования Pascal) 

 
uses crt; 
const mf=51; 
         eps=1e-5; 
         sigma=5.669e-8; 
type 
     vector1=array[1..mf] of real; 
     vector2=array[1..mf,1..mf] of real; 
var {раздел описания переменных, которые мы будем использовать в 
программе} 
 i, j, Nx, Ny, k, l                     : integer; 
 T, Ts, Tn                               : vector2; 
 alfa, beta                               : vector1; 
 ai, bi, ci, fi, max                    : real; 
 lamda, ro, c, eps1, kapa        : real; 
 hx, hy, tau, t_end, time         : real; 
 T0, Te, Lx, Ly, h, Th, Tc      : real; 
 f, g                                         : text; 
begin 
 clrscr; 
{с клавиатуры вводим все необходимые входные параметры} 
 Writeln('Введите количество пространственных узлов в пластине по оси 
х, Nx'); 
 Readln(Nx); 
 Writeln('Введите количество пространственных узлов в пластине по оси 
y, Ny'); 
 Readln(Ny); 
 Writeln('Введите окончание по времени, t_end'); 
 Readln(t_end); 
 Writeln('Введите длину пластины, Lx'); 
 Readln(Lx); 
 Writeln('Введите ширину пластины, Ly'); 
 Readln(Ly); 
 Writeln('Введите толщину пластины, h'); 
 Readln(h); 
 Writeln('Введите коэффициент теплопроводности материала пластины, 
lamda'); 
 Readln(lamda); 
 Writeln('Введите плотность материала пластины, ro'); 

 Readln(ro); 
 Writeln('Введите теплоемкость материала пластины, c'); 
 Readln(c); 
 Writeln('Введите коэффициент теплообмена, kapa'); 
 Readln(kapa); 
 Writeln('Введите приведенную степень черноты, eps1'); 
 Readln(eps1); 
 Writeln('Введите температуру окружающей среды, Te'); 
 Readln(Te); 
 Writeln('Введите температуру на границе x = 0, Th'); 
 Readln(Th); 
 Writeln('Введите температуру на границе x = Lx, Tc'); 
 Readln(Tc); 
 Writeln('Введите начальную температуру, T0'); 
 Readln(T0); 
{определяем расчетные шаги сетки по пространственным 
координатам} 
 hx:=Lx/(Nx-1); 
 hy:=Ly/(Ny-1); 
{определяем расчетный шаг сетки по времени} 
 tau:=t_end/100.0; 
{записываем в файл входные параметры для проверки} 
 Assign(f,'res.txt'); 
 Rewrite(f); 
 Writeln(f,'Длина пластины Lx = ',Lx:6:4); 
 Writeln(f,'Ширина пластины Ly = ',Ly:6:4); 
 Writeln(f,'Толщина пластины h = ',h:6:4); 
 Writeln(f,'Число узлов по пространственной координате x в пластине 
Nx = ',Nx); 
 Writeln(f,'Число узлов по пространственной координате y в пластине 
Ny = ',Ny); 
 Writeln(f,'Коэффициент теплопроводности материала пластины lamda = 
',lamda:6:4); 
 Writeln(f,'Плотность материала пластины ro = ',ro:6:4); 
 Writeln(f,'Теплоемкость материала пластины с = ',c:6:4); 
 Writeln(f,'Коэффициент теплообмена kapa = ',kapa:6:4); 
 Writeln(f,'Приведенная степень черноты eps1 = ',eps1:6:4); 
 Writeln(f,'Начальная температура T0 = ',T0:6:4); 
 Writeln(f,'Температура окружающей среды Te = ',Te:6:4); 
 Writeln(f,'Температура на границе x = 0 Th = ',Th:6:4); 
 Writeln(f,'Температура на границе x = Lx Tc = ',Tc:6:4); 

 Writeln(f,'Результат получен с шагом по координате x hx = ',hx:6:4); 
 Writeln(f,'Результат получен с шагом по координате y hy = ',hy:6:4); 
 Writeln(f,'Результат получен с шагом по времени tau = ',tau:6:4); 
 Writeln(f,'Температурное поле в момент времени t = ',t_end:6:4); 
 close(f); 
{определяем поле температуры в начальный момент времени} 
 for i:= 1 to Nx do 
  for j:= 1 to Ny do 
    T[i,j]:=T0; 
{проводим интегрирование нестационарного уравнения 
теплопроводности} 
 time:=0; 
 while time<t_end do {используем цикл с предусловием} 
  begin 
{увеличиваем переменную времени на шаг τ} 
   time:=time+tau; 
{запоминаем поле температуры на предыдущем n временном слое} 
   for i:=1 to Nx do 
    for j:=1 to Ny do 
      Tn[i,j]:=T[i,j]; 
{решаем СЛАУ в направлении оси Ох для определения поля 
температуры на промежуточном (n+1/2) временном слое} 
   for j:=1 to Ny do 
    begin 
{определяем начальные прогоночные коэффициенты на основе левого 
граничного условия} 
     alfa[1]:=0.0; 
     beta[1]:=Th; 
{цикл с постусловием, позволяющий итерационно вычислять значения 
температуры в области решения, вследствие наличия нелинейности в 
самом уравнении теплопроводности} 
     repeat 
{запоминаем поле температуры на предыдущей s итерации} 
      for k:=1 to Nx do 
       for l:=1 to Ny do 
        Ts[k,l]:=T[k,l]; 
{цикл с параметром для определения прогоночных коэффициентов по 
формуле (8)} 
      for i:= 2 to Nx-1 do 
       begin 

{ai, bi, ci, fi – коэффициенты канонического представления систему 
уравнений с трехдиагональной матрицей} 
        ai:=lamda/sqr(hx); 
        bi:=2.0*lamda/sqr(hx)+kapa/h+ro*c/tau; 
        ci:=lamda/sqr(hx); 
        fi:=-ro*c*Tn[i,j]/tau-kapa*Te/h-eps1*sigma 
              *(sqr(sqr(Te))-sqr(sqr(T[i,j])))/h; 
{alfa[i], beta[i] – прогоночные коэффициенты} 
        alfa[i]:=ai/(bi-ci*alfa[i-1]); 
        beta[i]:=(ci*beta[i-1]-fi)/(bi-ci*alfa[i-1]); 
       end; 
{определяем значение температуры на правой границе на основе 
правого граничного условия} 
      T[Nx,j]:=Tc; 
{используя соотношение (7) определяем неизвестное поле температуры 
на промежуточном (n+1/2) временном слое} 
      for i:= Nx-1 downto 1 do 
       T[i,j]:=alfa[i]*T[i+1,j]+beta[i]; 
{определяем максимум модуля разности локальных значений 
температур на текущей и на предыдущей итерации} 
      max:=abs(T[1,j]-Ts[1,j]); 
      for i:=2 to Nx do 
       if max < abs(T[i,j]-Ts[i,j]) then max:=abs(T[i,j]-Ts[i,j]); 
     until max<=eps; {окончание цикла с постусловием} 
    end; {поле температуры на промежуточном (n+1/2) временном слое 
определили} 
{запоминаем поле температуры на промежуточном (n+1/2) временном 
слое} 
   for i:=1 to Nx do 
    for j:=1 to Ny do 
      Tn[i,j]:=T[i,j]; 
{решаем СЛАУ в направлении оси Оу для определения поля 
температуры на целом (n+1) временном слое} 
   for i:=2 to Nx-1 do 
    begin 
{определяем начальные прогоночные коэффициенты на основе нижнего 
граничного условия, используя соотношения (20) при условии, что         
q1 = 0} 
     alfa[1]:=2.0*lamda*tau/(ro*c*sqr(hy)+2.0*lamda*tau); 
     beta[1]:=ro*c*sqr(hy)*Tn[i,1]/(ro*c*sqr(hy)+2.0*lamda*tau); 

{цикл с постусловием, позволяющий итерационно вычислять значения 
температуры в области решения, вследствие наличия нелинейности в 
самом уравнении теплопроводности} 
     repeat 
{запоминаем поле температуры на предыдущей s итерации} 
      for k:=1 to Nx do 
       for l:=1 to Ny do 
        Ts[k,l]:=T[k,l]; 
{цикл с параметром для определения прогоночных коэффициентов по 
формуле (8)} 
      for j:= 2 to Ny-1 do 
       begin 
{ai, bi, ci, fi – коэффициенты канонического представления систему 
уравнений с трехдиагональной матрицей} 
        ai:=lamda/sqr(hy); 
        bi:=2.0*lamda/sqr(hy)+kapa/h+ro*c/tau; 
        ci:=lamda/sqr(hy); 
        fi:=-ro*c*Tn[i,j]/tau-kapa*Te/h-eps1*sigma 
             *(sqr(sqr(Te))-sqr(sqr(T[i,j])))/h; 
{alfa[j], beta[j] – прогоночные коэффициенты} 
        alfa[j]:=ai/(bi-ci*alfa[j-1]); 
        beta[j]:=(ci*beta[j-1]-fi)/(bi-ci*alfa[j-1]); 
       end; 
{определяем значение температуры на верхней границе, используя 
соотношение (21) при условии, что q2 = 0} 
      T[i,Ny]:=(2.0*lamda*tau*beta[Ny-1]+ro*c*sqr(hy)*Tn[i,Ny]) 
                     /(ro*c*sqr(hy)+2.0*lamda*tau*(1.0-alfa[Ny-1])); 
{используя соотношение (7) определяем неизвестное поле температуры 
на целом (n+1) временном слое} 
      for j:= Ny-1 downto 1 do 
       T[i,j]:=alfa[j]*T[i,j+1]+beta[j]; 
{определяем максимум модуля разности локальных значений 
температур между текущей и предыдущей итерациями} 
      max:=abs(T[i,1]-Ts[i,1]); 
      for j:=2 to Ny do 
       if max < abs(T[i,j]-Ts[i,j]) then max:=abs(T[i,j]-Ts[i,j]); 
     until max<=eps; {окончание цикла с постусловием} 
    end; {поле температуры на целом (n+1) временном слое определили} 
  end; {цикл с предусловием окончен} 
{выводим результат в файл} 
 Assign(g,'tempr.txt'); 

 Rewrite(g); 
 for i:=1 to Nx do 
  for j:=1 to Ny do 
   writeln(g,'  ',hx*(i-1):10:8,'  ',hy*(j-1):10:8,'   ',T[i,j]:8:5); 
 close(g); 
end. 
 
 Получено следующее распределение температуры: 
 

 
Рис. 31. Распределение температуры (ºС) в пластине при t = 3600 с 

